


[bookmark: _GoBack]/*************************************************************************
*   This program will calculate the recodes for the publication tables.
*
*   Program:      tbl_rcd_main.sas
*   Programmed by: Diane Schwartz
*
*   Input directories:  /hspb/<survey year>/recodes/proddata
*                       /hspb/<survey year>/edits/proddata
*                       /hspb/<survey year/serialsuffix/proddata
*
*   Output directories:   /hspb/<survey year>/recodes/proddata/tbl
*                         /hspb/<survey year>/recodes/proddata
*
** Each section with the exception of the last two, are separate programs
*
* Variables that are calculated are:
* Section One -- persons rostered information
*    zinc -- income of ref. person and related hh members
*    salinc -- salary income of ref.person and related hh members
*    zinc2 -- total household income
*    zadult -- number of adults 18 and over
*    hhldkid -- number of householders children ( any age)
*    hhkdu18 -- number of houeholders children under 18
*    skdu18 -- number of single kids under 18
*    skidu6 -- number of single kids under 6
*    skd617 -- number of single kids ages 6 to 17
*    adltkd29 -- number of adult householders children,
*                ages 18 to 29 who are not married
*    adltkd30 -- number of adult householders children,
*                ages 30 or more who are not married
*    kid013 -- number of children in household between ages 0 and 13
*    kid515 -- number of children in household between ages 5 and 15
*    kid -- number of children in household
*    kidu6 -- number of children under 6 year of age and single
*    kid617 -- number of children ages 6 to 17 who are single
*    elder -- number of elderly persons ages 65 and older
*    senior -- number of persons age 55 and older
*    hhage -- householders age
*    hhsex -- householders sex
*    hhspan -- householders spanish origin
*    hhrace -- householders race
*    hhmove -- householders move in date
*    hhmar -- householders marital status
*    hhgrad -- householders educational level
*    hhrel -- householders relationship code
*    hhcit -- householders citizenship code
*    hhinusyr -- householders immigration year
*    spouse -- spouse of householder is present
*    nonrel -- number of nonrelatives in household
*    coown -- coowner
*    nrldgr -- non-related lodger
*    unrlkid -- unrelated kid ages 0 to 17
*    unrlper -- unrelated person 18 and over
*    parent -- parent of householder present
*    pvother -- individuals wages and salaries
*    grandpa -- grandparent of householder present
*    grankid -- grandchild of householder present
*    refper -- reference person flag
*    relfam -- number of related subfamilies
*    relhhage -- related subfamilies head of families age
*    reloth --  households with other types of relatives
*    othrel -- other relatives of householder
*    nrel2 -- 2 person household - not related
*    nrel3 -- 3-8 person household - not related
*    norel3 -- count of not related household members
*    scndry -- number of secondary families - not related to householder
*    respdnt -- household respondent that moved in the last year
*    hrmar -- household respondents marital status
*    movedly -- flag for householder moved within the last year
*    vother2 -- total non-wage income for household
*
*  Section Two -- main portion of recodes
*    phone  --  phone
*    kitchen -- complete kitchen
*    rooms -- number of rooms
*    confeet - confeetr -- rounded at end to be confeet -- condominium and
*                          cooperative fee
*    zcrowf - zcrowfr -- rounded at end to be zcrowf -- square feet per
*                        person
*    zcrowd - zcrowdr -- rounded at end to be zcrowd -- persons per room
*    zcrowb - zcrowbr -- rounded at end to be zcrowb -- persons per bedroom
*    tprice -- purchase price
*    PCTDWN -- Percent Down - added in 2007
*    zaq1 -- tally for upkeep in selected physical problems
*    zadeq -- tally to determine severe or moderated selected physical problems
*    maint - maintr -- rounded at end to be maint -- routine maintenance in
*                      last year
*    zsmhcn - zsmhcnr -- rounded at end to be zsmhcn -- monthly housing costs
*                        withour maintenance
*    zsmhcm - zsmhcmr -- rounded at end to be zsmchm -- monthly housing costs
*                        with maintenance
*    zsmhco - zsmhcor -- rounded at end to be zsmhco -- monthly housing costs
*                        excluding 2nd and subsequent mortgages and maintenance
*    zsmhcp - zsmhcpr -- rounded at end to be zsmhcp -- monthly housing costs
*                        as a percent of current income
*    zvi - zvir -- rounded at end to be zvi -- ratio of value to current income
*    amtomo - amtomor -- rounded at end to be amtomo -- average monthly costs
*                        for fuel oil
*    amtimo - amtimor -- rounded at end to be amtimo -- average monthly costs
*                        for property insurance
*    amtwmo - amtwmor -- rounded at end to be amtwmo -- average monthly costs
*                        for water
*    amttmo - amttmor -- rounded at end to be amttmo -- average monthly costs
*                        for trash
*    amtgmo - amtgmor -- rounded at end to be amtgmo -- average monthly costs
*                        for bottled gas
*    amtfmo - amtfmor -- rounded at end to be amtfmo -- average monthly costs
*                        for other fuel
*    lrntmo - lrntmor -- rounded at end to be lrntmo -- land rent fee paid
*    amtxp - amtxpr -- rounded at end to be amtxp -- average monthly costs
*                      for real estate taxes
*    amtxv - amtxvr -- rounded at end to be amtxv -- annual taxes paid per
*                      $1000 value
*    newc -- new construction
*    pubsub -- rent reductions
*    saving -- savings -- no longer being used as of N2005
*
*  Section Three -- poverty lookup table
*    poor - poorr -- rounded at end to be poor -- household income as percent
*                    of poverty level
*    pvlkup -- not used in tables, only for verification -- key to access the
*              poverty lookup table
*    povlvl -- not used in tables, only for verification -- value in poverty
*              lookup table corresponding to lookup key
*
*  Section Four -- calculations for INCM05 stub
*    salinc  -- wages and salary
*    inc3flg -- flag determining if 2 or more people each earned over 20% of
*               wages and salary
*    per20wg -- not used in tables, only for verification -- 20% wage level
*
*  Section Five -- first part of the mortgage recodes
*    unpbalt - unpbaltr -- rounded at end to be unpbalt -- total 
*               outstanding principle
*    clpeva - clpevar -- rounded at end to be clpeva -- current load as percent
*                        of value
*    crintr - crintrr -- rounded at end to be crintr -- current interest rate
*    pmtt -- monthly mortgage payment
*
*  Section Five-a -- another mortgage recode
*    pmtin -- monthly mortgage payment as percent of income
*
*  Section Six -- last part of the mortgage recodes
*    rymort -- remaining years mortgaged
*    prin01 - prin01r -- rounded at end to be prin01 -- monthly payment for
*                        principle and interest
*    paypm1 -- payment plan for primary mortgage
*    paypm2 -- payment plan for secondary mortgage
*    credlp -- type of home equity mortgages (no longer using for n2001)
*    crdlmt -- total home-equity line-of-credit limit
*    crdlns -- total outstanding line-of-credit loans
*    crdint -- current line-of-credit interest rate
*    crdpmt -- line-of-credit monthly payment
*
*  Section Seven -- recent mover recode
*    amvone -- household all moved from one unit
*    mv1hh -- householder of previous unit status of moving from one unit
*    mv2hh -- householder of previous unit status of moving from 2+ units
*    amvtwo -- houshold moved from 2+ units
*    smv3 -- some already here, rest moved in
*    mv3hh -- householder of previous unit status of moving when some were here
*    mv4nr -- previous units not reported
*    lastyr -- total with a move in the last year
*    xheadres -- respondents xhead value
*    xunitres -- respondents xunit value
*    xcostres -- respondents xcost value
*    xperres -- respondents xper value
*    xinusres -- respondents xinus value
*    xtenres -- respondents xten value
*    xrelres -- respondents xrel value
*    rmvunvres -- recent movers universe for the respondent
*
*  Section Eight -- journey to work recodes  (not done in 2011)
*    j1drv -- drives self
*    j1cp -- carpool
*    j1cp2 -- carpool - 2 person
*    j1cp3 -- carpool - 3 person
*    j1cp4 -- carpool - 4+ person
*    j1mass -- mass transportation
*    j1taxi -- taxicab
*    j1bike --  bicycle or motorcycle
*    j1walk -- walks only
*    j1oth -- other meansa
*    j1home -- works at home
*    j2u15 --  under 15 minutes
*    j21529 -- 15 to 29 minutes
*    j23044 -- 30 to 44 minutes
*    j24559 -- 45 to 59 minutes
*    j26089 -- 60 to 89 minutes
*    j2o90 -- over 90 minutes
*    j2home -- works at home
*    j2noplc -- no fixed place of work
*    j3u1 -- under 1 mile
*    j314 -- 1 to 4 miles
*    j359 -- 5 to 9 miles
*    j31019 -- 10 to 19 miles
*    j32029 -- 20 to 29 miles
*    j33049 -- 30 to 49 miles
*    j3o50 -- over 50 miles
*    j3home -- works at home
*    j3noplc -- no fixed place of work
*    j4123a -- 12 to 3 am
*    j436a -- 3 to 6 am
*    j467a -- 6 to 7 am
*    j47730a -- 7 to 7:30 am
*    j47308a -- 7:30 to 8 am
*    j48830a -- 8 to 8:30 am
*    j48309a -- 8:30 to 9 am
*    j4910a -- 9 to 10 am
*    j4104p -- 10 am to 4 pm
*    j4412p -- 4 to 12 pm
*    j4nr -- departure time not reported
*    j9whtot -- total people who worked at home
*    j9wh19 -- hours worked at home between 1 and 9
*    j9wh1019 -- between 10 and 19
*    j9wh2029 -- between 20 and 29
*    j9wh3039 -- between 30 and 39
*    j9wh40pl -- 40 or more
*    j9whnr -- work hours not reported
*    j9dy0 -- days worked at home instead of travelling is 0
*    j9dy12 -- 1-2 days
*    j9dy34 -- 3-4 days
*    j9dy5pl -- 5 or more days
*    j9dynr -- days not reported
*    j9nowrk -- did not work at home
*    j9nr -- worked at home not reported
*    hhwlineq -- housholders wlineq value
*    hhtran -- householders tran value
*    hhvehcl -- householders vehcl value
*    hhpass -- householders pass value
*    hhhjob -- householders job value
*    hhtimej -- householders timej value
*    hhdistj -- householders distj value
*    hhampm -- householders ampm value
*    hhwtime -- householders wtime value
*    hhwhome -- householders whome value
*    hhwhhrb -- householders whhrb value
*    hhwhhrw -- householders whhrw value
*    hhwhdy -- householders whdy value
*    jtwtot -- total persons that worked last week
*    hhwrkhom -- householder worked at home
*    hhhrswrk -- hours householder worked at home
*
*  Section Nine -- RENT01 - lodgers recode and spouses grandparents recodes
*    lodg - lodgr -- rounded at end to be lodg -- rent paid by lodgers
*    ldgrs -- flag indicating it was a lodger
*    sgrndpa -- spouses grandparent found -- used in calculating gen3 for COMP05
*    sprntln -- spoused parents line number
*
*** the next two sections are part of the main program.
*  Section Eleven -- Combine previous data sets and do all rounding
*    confeet
*    zcrowf
*    zcrowd
*    zcrowb
*    zsmhcn
*    zsmhcm
*    zsmhco
*    zsmhcv
*    zsmhcp
*    zvi
*    amtomo
*    amtimo
*    amtwmo
*    amttmo
*    amtgmo
*    amtfmo
*    lrntmo
*    amtxp
*    amtxv
*    maint
*    poor
*    clpeva
*    crintr
*    prin01
*    lodg
*
*  Section Twelve -- final section -- resets missing values in rounded
*                    variables and calculates gen3
*    gen3 -- three generations live in household -- COMP05
*
****************************************************************************/

options mprint symbolgen mlogic;

libname edt "$INPDR";
libname undp "$UNDP";
libname rcd "$INPDR2";
libname rcds "$RECDS";
libname rcdp "$RECDS";  /* used in both puf and tbl processing */
libname temp "$TEMP";
libname mstr "$MSTR";

libname library "$INPDR2";

/* ****  variables passed from script *** */
%global survyear msrtkey;
%let survyear =%sysget (YRSURV);
%let msrtkey=%sysget (MSRTKEY);

missing D R B X M N U _ S ;

/* ************************************************************************ */
/*               Prologue section                                           */
/*                                                                          */
/*       Sort datasets prior to using them,  For 2007 the edits are run     */
/*       in the national format and are not in the correct format for MS    */
/*       runs.  This section will make sure that all datasets are ready     */
/*       when they are needed!                                              */
/* ************************************************************************ */

  proc sort data=edt.houshld out=houshld;
    by &msrtkey;
  run;

  proc sort data=rcd.person out=person;
    by &msrtkey;
  run;

  proc sort data=undp.houshldr out=uhoushldr;
     by &msrtkey;
  run;

/*  proc sort data=edt.jtw out=jtw;
     by &msrtkey;
  run;  */

  proc sort data=edt.mortg out=mortg;
     by &msrtkey;
  run;

  proc sort data=edt.rmov out=rmov;
     by &msrtkey movgrp;
  run;

  proc sort data=mstr.master out=master;
     by &msrtkey;
  run;

/* *************************************************************************
*             S E C T I O N   O N E
*
*      This section creates a data set with the recodes included from the
*       persons data set
*************************************************************************** */

%include "$PROG/sect1_per_inc.sas"/source2;

/* *************************************************************************
*             S E C T I O N   T W O
*
*      Now do the processing not involving the persons data set or any of
*       the other rostered data sets!
*************************************************************************** */

%include "$PROG/sect2_hh_inc.sas"/source2;

/* *************************************************************************
*             S E C T I O N   T H R E E
*
*      Calculate POOR by using a poverty look up table
*************************************************************************** */

%include "$PROG/sect3_poor_inc.sas"/source2;

/* *************************************************************************
*             S E C T I O N   F O U R
*
*      This section calculates the inc3flg for the INCM05 stub
*************************************************************************** */

%include "$PROG/sect4_incm05_inc.sas"/source2;

/* *************************************************************************
*             S E C T I O N   F I V E
*
*     The first part of the mortgage variable recodes
*************************************************************************** */

%include "$PROG/sect5_mortg_pt1_inc.sas"/source2;

/* *************************************************************************
*            S E C T I O N   F I V E - A
*
*    For 2011 a new variable needs perrcd and mtgrcd to calculate PMTINR
**************************************************************************** */

%include "$PROG/sect5a_mortg_inc.sas"/source2;

/* *************************************************************************
*             S E C T I O N   S I X
*
*      The second part of mortgage variable recodes
*************************************************************************** */

%include "$PROG/sect6_mortg_pt2_inc.sas"/source2;

/* *************************************************************************
*             S E C T I O N   S E V E N
*
*     Recent movers recode for table stub MOVE02
*************************************************************************** */

%include "$PROG/sect7_move02_inc.sas"/source2;

/* *************************************************************************
*       S E C T I O N   S E V E N – A and E I G H T not run for 
*                           2011, not included 
*/
/* *************************************************************************
*             S E C T I O N   N I N E
*
*      Recode for lodg and ldgrs
*************************************************************************** */

%include "$PROG/sect9_lodg_inc.sas"/source2;

/* *************************************************************************
*             S E C T I O N   T E N
*
*      Recode for lodg and ldgrs
*************************************************************************** */

%include "$PROG/sect10_sbfixu_inc.sas"/source2;


/* *************************************************************************
*             S E C T I O N   E l E V E N
*
*      Merge all previous data sets together and do the rounding on the
*       calculated variables
*************************************************************************** */

data rcds.tblone;

length CONFEET 3. ZCROWF 3. ZCROWD 3. ZCROWB 3. ZSMHCN 4. ZSMHCM 4. ZSMHCO 4.
       ZSMHCP 4. ZVI 4. AMTOMO 3. AMTIMO 3. AMTWMO 3. AMTTMO 3.
       AMTGMO 3. AMTFMO 3. LRNTMO 3. AMTXP 4. AMTXV 4. MAINT 4. POOR 4.
       CLPEVA 4. CRINTR 4. PRIN01 4. LODG 4. PCTDWN 4. PMTIN 4.;

   merge rcds.perrcd
         rcds.hhrcd
         rcds.povrcd
         rcds.inc3rcd
         rcds.mtgrcd
         rcds.pmtinrcd
         rcds.mtg2rcd
         rcds.mv2rcd
    /*   rcds.jtwrcd   * not used in 2011  */
         rcds.rnt01rc
    /*     rcds.msgeoars (keep=&msrtkey subarea)  * only use for metro */
    ;
   by &msrtkey;

   if (confeetr ne . and confeetr ne .B) then
      confeet = round(confeetr, 1);
   if (zcrowfr ne . and zcrowfr ne .B) then
     zcrowf = round(zcrowfr,1);
   if (zcrowdr ne . and zcrowdr ne .B) then
     zcrowd = round(zcrowdr,.01);
   if (zcrowbr ne . and zcrowbr ne .B) then
     zcrowb = round(zcrowbr,.01);
   if (zsmhcnr ne . and zsmhcnr ne .B) then
     zsmhcn = round(zsmhcnr,1);
   if (zsmhcmr ne . and zsmhcmr ne .B) then
     zsmhcm = round(zsmhcmr,1);
   if (zsmhcor ne . and zsmhcor ne .B) then
     zsmhco = round(zsmhcor,1);
   if (zsmhcpr ne . and zsmhcpr ne .B and zsmhcpr ne .X) then
     zsmhcp = round(zsmhcpr,1);
   if (zvir ne . and zvir ne .B) then
     zvi = round(zvir, .01);
   if (amtomor ne . and amtomor ne .B) then
     amtomo = round(amtomor,1);
   if (amtimor ne . and amtimor ne .B) then
     amtimo = round(amtimor,1);
   if (amtwmor ne . and amtwmor ne .B) then
     amtwmo = round(amtwmor,1);
   if (amttmor ne . and amttmor ne .B) then
     amttmo = round(amttmor,1);
   if (amtgmor ne . and amtgmor ne .B) then
     amtgmo = round(amtgmor,1);
   if (amtfmor ne . and amtfmor ne .B) then
     amtfmo = round(amtfmor,1);
   if (lrntmor ne . and lrntmor ne .B) then
     lrntmo = round(lrntmor,1);
   if (amtxpr ne . and amtxpr ne .B) then
     amtxp = round (amtxpr,1);
   if (amtxvr ne . and amtxvr ne .B) then
     amtxv = round(amtxvr,1);
   if (maintr ne . and maintr ne .B and
       maintr ne .D and maintr ne .R) then
     maint = round(maintr,1);
   if (poorr ne . and poorr ne .B) then
     poor = round(poorr,1);
   if (clpevar ne .  and clpevar ne .B) then
     clpeva = round(clpevar,1);
   if (crintrr ne . and crintrr ne .B) then
     crintr = round(crintrr,.1);
   if (prin01r ne . and prin01r ne .B) then
     prin01 = round(prin01r,1);
   if (lodgr ne . and lodgr ne .B and lodgr ne .D) then
     lodg = round(lodgr,1);
   if (PCTDWNR ne . and PCTDWNR ne .B and PCTDWNR ne .D) then
     PCTDWN = round(PCTDWNR,1);
   if (PMTINR ne . and PMTINR ne .B and PMTINR ne .D) then
     PMTIN =  round(PMTINR,1);

run;
/* ************************************************************************
*              S E C T I O N  E L E V E N - A
*
*    This section pulls the variable SURVEYCD from the masterfile and puts it
*    on the table recode dataset
*
************************************************************************** */
data tblone_a;
   merge rcds.tblone (in=tblin)
         master (keep=&msrtkey surveycd);
   by &msrtkey;

   if tblin then
      output;
run;


/* **************************************************************************
*              S E C T I O N   T W E L V E
*
*    Final section   ---    reset rounded variables to special missings if they
*                         were before rounding and calculate gen3 used in the
*                         COMP05 table stub.
* ************************************************************************** */
data tblrcd ;

length GEN3 $1;


    set tblone_a;

     if confeet = . then
        confeet = confeetr;

     if zcrowf = . then
        zcrowf = zcrowfr;

     if zcrowd = . then
        zcrowd = zcrowdr;

     if zcrowb = . then
        zcrowb = zcrowbr;

     if zsmhcn = . then
        zsmhcn = zsmhcnr;

     if zsmhcm = . then
        zsmhcm = zsmhcmr;

     if zsmhco = . then
        zsmhco = zsmhcor;

     if zsmhcp = . then
        zsmhcp = zsmhcpr;

     if zvi = . then
        zvi = zvir;

     if amtomo = . then
        amtomo = amtomor;
     if amtomo = 0 then
        amtomo = 1;

     if amtimo = . then
        amtimo = amtimor;
     if amtimo = 0 then
        amtimo = 1;

     if amtwmo = . then
        amtwmo = amtwmor;
     if amtwmo = 0 then
        amtwmo = 1;

     if amttmo = . then
        amttmo = amttmor;
     if amttmo = 0 then
        amttmo = 1;

     if amtgmo = . then
        amtgmo = amtgmor;
     if amtgmo = 0 then
        amtgmo = 1;

     if amtfmo = . then
        amtfmo = amtfmor;
     if amtfmo = 0 then
        amtfmo = 1;

     if lrntmo = . then
        lrntmo = lrntmor;

     if amtxp = . then
        amtxp = amtxpr;

     if amtxv = . then
        amtxv = amtxvr;

     if maint = . then
        maint = maintr;

     if poor = . then
        poor = poorr;

     if clpeva = . then
        clpeva = clpevar;

     if crintr = . then
        crintr = crintrr;

     if prin01 = . then
        prin01 = prin01r;

     if lodg = . then
        lodg = lodgr;

     if PCTDWN = . then
        PCTDWN = PCTDWNR;
     if PCTDWN > 100 then
        PCTDWN = 100;

     if PMTIN = . then
        PMTIN = PMTINR;
      
/* gen3 recode -- COMP05 -- set gen3 if one of three possibilities exist */

   if (parent = 'X' or sparnt = 'X') and hhldkid > 0 and refper = 'X' then
        gen3 = 'X';

   if refper = 'X' and hhldkid > 0 and grankid = 'X' then
        gen3 = 'X';

   if refper = 'X' and (parent = 'X' or sparnt = 'X') and (grandpa = 'X'
      or sgrndpa = 'X') then
        gen3 = 'X';

data rcd.tblrcd;
  merge tblrcd 
        rcds.dbrcd (keep = &msrtkey hhear hsee hmemry hwalk hcare
                          herrnd hdsb withdis dkrfct)
        rcds.hmdrcd (keep = &msrtkey hkcab hcab hstove hctruse hgetbr
                          hsink hfaucet hbtub hwshwr hstoop hreach
                          hgrasp withdiff dkrfct 
                          rename=(dkrfct=difdkrfct))
       ; 
   by &msrtkey;

run;
 /* end of table recode program */


/***The following programs contain code that is included in the table recode programs.  The code either  contains macro code, or sections of code that have been called as a include portion in the previous  code or within the sections of code listed below.****/

/*  POVERTY_RCD_INCLUDE.SAS  */
%macro countkids;
  /* hhkdu18 recode -- used in POOR recode when accessing the poverty lookup
   table */

  if 0 <= AGE < 18  and REL > 2 then 
  do;
    HHKDU18+1;
    hhkdu18cnta+1;
  end;


  /* the following recodes all have to do with the children in the 
  household */

  if REL = 22 then 
  do;
      /* hhldkid recode -- used in setting the GEN3 recode */
      HHLDKID+1;
      hhldkidcnt+1;
  end;
%MEND countkids;


/* Calculating poverty */

%macro calcpov;

  /*  calculate the lookup variable to retrieve the poverty threshhold  */
  if (CURPER = 1 or CURPER = 2) then  
  do;

       PERS=put(CURPER,1.);
       perscnta+1;

       if 0 < = HHAGE < 65 then
       do;
         AGEHH='1';
         agehhcnt1+1;
       end;
       else if HHAGE ge 65 then 
       do;
         AGEHH='2';
         agehhcnt2+1;
       end;
  end;  
  else if CURPER > 2 then 
  do;

       AGEHH='3';
       agehhcnt3+1;

       if CURPER >= 9 then 
       do;      
           PERS='9';
           perscntb+1;
           HHKDU18 =  max(9-(CURPER-HHKDU18),0);
           hhkdu18cntb+1;
       end;

       else if CURPER < 9 then do;
           PERS=put(CURPER,1.);
           perscntc+1;
       end;
  end;

  if HHKDU18 ge 8 then
  do;
     KIDS='8';
     kidscnta+1;
  end;
  else if 0 <= HHKDU18 < 8 then
  do;
     KIDS=put(HHKDU18,1.);
     kidscntb+1;
  end;


  if 2 <= DATEM <= 6 then 
  do;
    IMONTH='06';
    imonthcnt6+1;
  end;
  else if DATEM = 11 then 
  do;
    IMONTH='11';
    imonthcnt11+1;
  end;
  else if DATEM = 12 then 
  do;
    IMONTH='12';  
    imonthcnt12+1;
  end;
  else if DATEM = 7 then 
  do;
    IMONTH='07';  
    imonthcnt7+1;
  end;
  else if DATEM = 8 then
  do;
    IMONTH='08';
    imonthcnt8+1;
  end;
  else if DATEM = 9 then 
  do;
    IMONTH='09';
    imonthcnt9+1;
  end;
  else if DATEM = 10 then 
  do;
    IMONTH='10'; 
    imonthcnt10+1;
  end;
  else if DATEM = 1 then
  do;
    IMONTH = '13';
    imonthcnt01+1;
  end;

  PVLKUP = IMONTH||PERS||AGEHH||KIDS;
  pvlkupcnt+1;

%MEND calcpov;

/*  inc_recpt_incl.sas  */
/*******************************************************************
*  New income recode variables for reference person and related    *
*  individuals (family level) for Receipt of ... income ...        *
* Created by: Colette Fladger                                      *
* Date:       December 2005                                        *
* Program name:  inc_recpt_incl.sas                                *
* Updated - D Knoll                                                *
*           3/2006                                                 *
*           added edit flags                                       *
*******************************************************************/
%macro recptinc;


       if (age > = 16 and rel < 30) then
         do;

/***  Receipt of self-employment income  ***/

              if (PQSELF = '1') then
                do;
                   QSELF = '1';
                   qslfcnt1+1;
                end;

              else
                do;
                   if (QSELF ne '1') then
                   do;
                     QSELF = '2';
                     qslfcnt2+1;
                   end;
                end;

              if (JPQSELF = '2' and JQSELF ne '2') then
                do;
                   JQSELF = '2';
                   jqslfcnt2+1;
                end;

              else
                do;
                   if (JPQSELF eq '1' and JQSELF ne '2') then
                   do;
                     JQSELF = '1';
                     jqslfcnt1+1;
                   end;
                end;


/***  Receipt of interest, dividends, net rental income, royalty ***/
/***  income or income from estates and trusts                   ***/
/*** the variable PQIDRI was replaced by 3 variables in n2007 ***/
/*  The first variables is QINT */
            if (PQINT eq '1') then
                do;
                   QINT = '1';
                   qintcnt1+1;
                end;

              else
                do;
                   if (QINT ne '1') then
                   do;
                     QINT = '2';
                     qintcnt2+1;
                   end;
                end;

              if (JPQINT = '2' and JQINT ne '2') then
                do;
                   JQINT = '2';
                   jqintcnt2+1;
                end;

              else
                do;
                   if (JPQINT eq '1' and JQINT ne '2') then
                   do;
                     JQINT = '1';
                     jqintcnt1+1;
                   end;
                end;
/* the second variable is QDIV */
            if (PQDIV eq '1') then
                do;
                   QDIV = '1';
                   qdivcnt1+1;
                end;

              else
                do;
                   if (QDIV ne '1') then
                   do;
                     QDIV = '2';
                     qdivcnt2+1;
                   end;
                end;

              if (JPQDIV = '2' and JQDIV ne '2') then
                do;
                   JQDIV = '2';
                   jqdivcnt2+1;
                end;

              else
                do;
                   if (JPQDIV eq '1' and JQDIV ne '2') then
                   do;
                     JQDIV = '1';
                     jqdivcnt1+1;
                   end;
                end;
/* the third variable is QRENT */
            if (PQRENT eq '1') then
                do;
                   QRENT = '1';
                   qrntcnt1+1;
                end;

              else
                do;
                   if (QRENT ne '1') then
                   do;
                     QRENT = '2';
                     qrntcnt2+1;
                   end;
                end;

              if (JPQRENT = '2' and JQRENT ne '2') then
                do;
                   JQRENT = '2';
                   jqrntcnt2+1;
                end;

              else
                do;
                   if (JPQRENT eq '1' and JQRENT ne '2') then
                   do;
                     JQRENT = '1';
                     jqrntcnt1+1;
                   end;
                end;

/***  Receipt of Social Security or railroad retirement benefits  ***/

              if (PQSS eq '1') then
                do;
                   QSS = '1';
                   qsscnt1+1;
                end;

              else
                do;
                   if (QSS ne '1') then
                   do;
                     QSS = '2';
                     qsscnt2+1;
                   end;
                end;

              if (JPQSS = '2' and JQSS ne '2') then
                do;
                   JQSS = '2';
                   jqsscnt2+1;
                end;

                else
                  do;
                     if (JPQSS eq '1' and JQSS NE '2') then
                     do;
                       JQSS = '1';
                       jqsscnt1+1;
                     end;
                end;

/***  Receipt of Supplemental Security Income  ***/

              if (PQSSI eq '1') then
                do;
                   QSSI = '1';
                   qssicnt1+1;
                end;

              else
                do;
                   if (QSSI ne '1') then
                   do;
                     QSSI = '2';
                     qssicnt2+1;
                   end;
                end;

              if (JPQSSI = '2' and JQSSI ne '2') then
                do;
                   JQSSI = '2';
                   jqssicnt2+1;
                end;

              else
                do;
                   if (JPQSSI eq '1' and JQSSI ne '2') then
                   do;
                     JQSSI = '1';
                     jqssicnt1+1;
                   end;
                end;

/***  Receipt of public assistance or welfare  ***/

             if (PQWELF eq '1') then
               do;
                  QWELF = '1';
                  qwlfcnt1+1;
               end;
             else
               do;
                  if (QWELF ne '1') then
                  do;
                    QWELF = '2';
                    qwlfcnt2+1;
                  end;
               end;

             if (JPQWELF = '2' and JQWELF ne '2') then
               do;
                  JQWELF = '2';
                  jqwlfcnt2+1;
               end;
             else
               do;
                  if (JPQWELF eq '1' and JQWELF ne '2') then
                  do;
                    JQWELF = '1';
                    jqwlfcnt1+1;
                  end;
               end;

/***  Receipt of retirement or survivors pensions  ***/

             if (PQRETIR eq '1') then
               do;
                  QRETIR = '1';
                  qretcnt1+1;
               end;

             else
               do;
                  if (QRETIR ne '1') then
                  do;
                    QRETIR = '2';
                    qretcnt2+1;
                  end;
               end;

             if (JPQRETIR = '2' and JQRETIR ne '2') then
               do;
                  JQRETIR = '2';
                  jqretcnt2+1;
               end;

             else
               do;
                  if (JPQRETIR  eq '1' and JQRETIR ne '2') then
                  do;
                    JQRETIR = '1';
                    jqretcnt1+1;
                  end;
               end;

/*** Receipt of SSDI, workers compensation, veterans disability, ***/
/*** other disability                                            ***/

             if (PQWKCMP eq '1') then
               do;
                  QWKCMP = '1';
                  qwkcmpcnt1+1;
               end;

             else
               do;
                  if (QWKCMP ne '1') then
                  do;
                    QWKCMP = '2';
                    qwkcmpcnt2+1;
                  end;
               end;

             if (JPQWKCMP = '2' and JQWKCMP ne '2') then
               do;
                  JQWKCMP = '2';
                  jqwkcmpcnt2+1;
               end;

             else
               do;
                  if (JPQWKCMP eq '1' and JQWKCMP ne '2') then
                  do;
                    JQWKCMP = '1';
                    jqwkcmpcnt1+1;
                  end;
               end;

/*** Receipt of VA payments, unemployment benefits, child support,***/
/*** alimony, other income                                        ***/
/*   The variable PQOTALM was replaced by 2 variables in n2007           ***/
/*  The first variable QALIM  */
             if (PQALIM eq '1') then
               do;
                  QALIM = '1';
                  qalmcnt1+1;
               end;

             else
               do;
                  if (QALIM ne '1') then
                  do;
                    QALIM = '2';
                    qalmcnt2+1;
                  end;
               end;

             if (JPQALIM = '2' and JQALIM ne '2') then
               do;
                  JQALIM = '2';
                  jqalmcnt2+1;
               end;

              else
                do;
                   if (JPQALIM eq '1' and  JQALIM ne '2') then
                   do;
                    JQALIM = '1';
                    jqalmcnt1+1;
                   end;
               end;
/*  the second variable QOTHER */
             if (PQOTHER eq '1') then
               do;
                  QOTHER = '1';
                  qothrcnt1+1;
               end;

             else
               do;
                  if (QOTHER ne '1') then
                  do;
                    QOTHER = '2';
                    qothrcnt2+1;
                  end;
               end;

             if (JPQOTHER = '2' and JQOTHER ne '2') then
               do;
                  JQOTHER = '2';
                  jqothrcnt2+1;
               end;

              else
                do;
                   if (JPQOTHER eq '1' and  JQOTHER ne '2') then
                   do;
                    JQOTHER = '1';
                    jqothrcnt1+1;
                   end;
               end;

  end;  /* end of Family level check of receipt of ... income  */



%MEND recptinc;

/********************************************************************
* MOB_ASST_INCL.SAS ...                                             *
********************************************************************/
/*******************************************************************
* Recoding the responses to the manufactured/mobile home questions *
* into one recode -5/06 dropped the last E on MOBILTYPE to MOBILTYP*
* as requested by Debbie Knoll for n2005                           *
*******************************************************************/
%macro mobil;
  if MHWIDE = 'B' then
  do;
    MOBILTYP = 'B';
    mobtypcntb+1;
  end;
  else if MHWIDE = '1' then 
  do;
    MOBILTYP = '1';
    mobtypcnt1+1;
  end;
  else if (MHWIDE = '2' or MHWIDE = '3') then 
  do;
    MOBILTYP = '2';
    mobtypcnt2+1;
  end;

%MEND mobil;
              
/**********************************************************************  
* The Eleven (11) assisted living services questions into two recodes *  
**********************************************************************/  
/*            ** 2011 these questions were not asked **
 %macro asstlvg;
  if (HOUSKP = 'B' and MEALS = 'B' and TRANSP = 'B' and
      FINMGMT = 'B' and PHONUSE = 'B' and SHOPNG = 'B') then
    do;
      ASSTSERV = 'B';
      asstservcntb+1;
    end;
  else if (HOUSKP = '1' or MEALS = '1' or TRANSP = '1' or
           FINMGMT = '1' or PHONUSE = '1' or SHOPNG = '1') then
    do;
      ASSTSERV = '1';
      asstservcnt1+1;
    end;
  else 
    do;
      ASSTSERV = '2';   
      asstservcnt2+1;
    end;
 
  if (BATHNG = 'B' or DRESNG = 'B' or EATNG = 'B' or 
      MOVNG = 'B' or TOLTUSE = 'B') then 
    do;
      PERSERV = 'B';
      perservcntb+1;
    end;
  else if (BATHNG = '1' or DRESNG = '1' or EATNG = '1' or 
      MOVNG = '1' or TOLTUSE = '1') then
    do;
      PERSERV = '1';
      perservcnt1+1;
    end;
  else 
    do;
      PERSERV = '2';  
      perservcnt2+1;
    end;

%MEND asstlvg;
*/

/*  sect1_per_inc.sas   */
/**************************************************************************
*   This program will calculate the recodes for the publication tables.
*
*   Program:      sect1_per_inc.sas
*   Programmed by: Diane Schwartz
*   Date:     January 31, 2012
*
* Variables that are calculated are:
* Section One -- persons rostered information
*    zinc -- income of ref. person and related hh members
*    salinc -- salary income of ref.person and related hh members
*    zinc2 -- total household income
*    zadult -- number of adults 18 and over
*    hhldkid -- number of householders children ( any age)
*    hhkdu18 -- number of householders children under 18 (unmarried)
*    hhldkidu18 -- number of householder children under 18
*    skdu18 -- number of single kids under 18
*    skidu6 -- number of single kids under 6
*    skd617 -- number of single kids ages 6 to 17
*    adltkd29 -- number of adult householders children,
*                ages 18 to 29 who are not married
*    adltkd30 -- number of adult householders children,
*                ages 30 or more who are not married
*    kid -- number of children in household
*    kidu6 -- number of children under 6 year of age and single
*    kidu5 -- number of children under 5 years of age
*    kid617 -- number of children ages 6 to 17 who are single
*    hhkid617 -- number of children ages 6 to 17 who are single or 
*                married
*    elder -- number of elderly persons ages 65 and older
*    hhage -- householders age
*    hhsex -- householders sex
*    hhspan -- householders spanish origin
*    hhrace -- householders race
*    hhmove -- householders move in date
*    hhmar -- householders marital status
*    hhgrad -- householders educational level
*    hhrel -- householders relationship code
*    hhcit -- householders citizenship code
*    hhinusyr -- householders immigration year
*    spouse -- spouse of householder is present
*    nonrel -- number of nonrelatives in household
*    coown -- coowner
*    nrldgr -- non-related lodger
*    unrlkid -- unrelated kid ages 0 to 17
*    unrlper -- unrelated person 18 and over
*    parent -- parent of householder present
*    vother -- individuals wages and salaries
*    grandpa -- grandparent of householder present
*    grankid -- grandchild of householder present
*    refper -- reference person flag
*    relfam -- number of related subfamilies
*    relhhage -- related subfamilies head of families age
*    reloth --  households with other types of relatives
*    othrel -- other relatives of householder
*    nrel2 -- 2 person household - not related
*    nrel3 -- 3-8 person household - not related
*    norel3 -- count of not related household members
*    scndry -- number of secondary families - not related to householder
*    respdnt -- household respondent that moved in the last year
*    hrmar -- household respondents marital status
*    movedly -- flag for householder moved within the last year
*    vother2 -- total non-wage income for household
*
****************************************************************************/

options mprint symbolgen mlogic;


/* ************************************************************************* */
/*                S E C T I O N    O N E                                     */
/*                                                                           */
/* ** this section creates a data set with the recodes included              */
/*    from the persons data set                                           ** */
/* ************************************************************************* */

/*** This include statement runs macros in:                             ***/
/*** Section One (countkids) and  Section Three (calcpov)               ***/
%include "$PROG/poverty_rcd_include.sas"/source2;

data rcds.perrcd (keep= &msrtkey ZADULT ZINC ZINC2 ELDER 
                      HHAGE SALINC SKDU18 KIDU6 KID617 HHKID617
                      ADLTKD29 ADLTKD30 SPOUSE RELOTH COOWN
                      RELFAM RELHHAGE NRLDGR UNRLKID UNRLPER
                      KID NONREL GRANDPA PARENT REFPER GRANKID
                      OTHREL HHCIT HHINUSYR HHSEX HHRACE
                      HHSPAN SCNDRY SKIDU6 SKD617 HHMAR NREL2
                      NREL3 HHGRAD HHMOVE HHREL HHLDKID RESPDNT
                      HRMAR MOVEDLY HHKDU18 STATUS CURPER DATEM
                      HHBLACK VOTHER VOTHER2 MOBILTYPE /* ASSTSERV
                      PERSERV */ QSELF QINT QDIV QRENT QSS QSSI QWELF
                      QRETIR QWKCMP QOTHER QALIM JSAL ctrlnum SAL
                      KIDU5 HHLDKIDU18
                      );

    length ZADULT 3. ZINC 5. ZINC2 5. ELDER 3.
           KIDU6 3. KID617 3. HHKID617 3. ADLTKD29 3. ADLTKD30 3.
           SKDU18 3. HHAGE 3. KIDU5 3. SALINC 5.
           SPOUSE $1 RELFAM 3. RELOTH 3. RELHHAGE 3. OTHREL 3.
           COOWN 3. NRLDGR 3. UNRLKID 3. UNRLPER 3. KID 3. NONREL 3.
           GRANDPA $1 PARENT $1 REFPER $1 
           GRANKID $1 HHSEX $1 HHRACE $2 HHCIT $1 HHINUSYR 4.
           HHSPAN $1  SCNDRY 3. SKIDU6 3. SKD617 3. HHMAR $1
           NREL2 $1 NREL3 $1 NOREL3 3. HHGRAD 3. HHMOVE 4. HHREL 3.
           HHLDKID 3. RESPDNT $1 HRMAR $1 MOVEDLY $1 HHKDU18 3.
           HHBLACK $1 VOTHER 5. VOTHER2 5. MOBILTYPE $1
           /* ASSTSERV $1 PERSERV $1 */ QSELF $1 QINT $1 QSS $1
           QSSI $1 QWELF $1 QRETIR $1 QWKCMP $1 QOTHER $1
           JSAL $1 QDIV $1 QRENT $1 QALIM $1 HHLDKIDU18 3.;

      merge person
            houshld (keep= &msrtkey status curper datey datem dline1
                               MHWIDE 
                               ctrlnum
                         ) end=send;
      by &msrtkey;

    retain incnt 0;
    retain ctrlno 0;
    retain ctllst 0;
    retain outcnt 0;
    retain zinc 0;
    retain salinc 0;
    retain zinc2 0;
    retain zadult 0;
    retain hhldkid 0;
    retain hhkdu18 0;
    retain hhldkidu18 0;
    retain skdu18 0;
    retain skidu6 0;
    retain skd617 0;
    retain adltkd29 0;
    retain adltkd30 0;
    retain kid 0;
    retain kidu6 0;
    retain kidu6 0;
    retain kid617 0;
    retain hhkid617 0;
    retain elder 0;
    retain hhage 0;
    retain hhsex ' ';
    retain hhspan ' ';
    retain hhrace ' ';
    retain hhmove 0;
    retain hhmar ' ';
    retain hhgrad 0;
    retain hhrel 0;
    retain hhcit ' ';
    retain hhinusyr 0;
    retain spouse ' ';
    retain nonrel 0;
    retain coown 0;
    retain nrldgr 0;
    retain unrlkid 0;
    retain unrlper 0;
    retain parent ' ';
    retain grandpa ' ';
    retain grankid ' ';
    retain refper ' ';
    retain relfam 0;
    retain relhhage 0;
    retain reloth 0;
    retain othrel 0;
    retain nrel2 ' ';
    retain norel3 0;
    retain nrel3 ' ';
    retain scndry 0;
    retain respdnt ' ';
    retain hrmar ' ';
    retain movedly ' ';
    retain cntrace 0;
    retain vother 0;
    retain vother2 0;
    retain MOBILTYPE ' ';
    retain QSELF ' ';
    retain QINT ' ';
    retain QDIV ' ';
    retain QRENT ' ';
    retain QSS ' ';
    retain QSSI ' ';
    retain QWELF ' ';
    retain QRETIR ' ';
    retain QWKCMP ' ';
    retain QOTHER ' ';
    retain QALIM ' ';
    retain jsal ' ';

  incnt+1;
/*    if(mod(incnt,500) = 0) then put 'processing obs number ' incnt; */

    if last.ctrlnm13 then ctllst+1;

    if first.ctrlnm13 then ctrlno+1;

 if first.ctrlnm13 then
 do;
   zinc = .B;
   salinc = .B;
   zinc2 = .B;
   zadult = .B;
   hhldkid = .B;
   hhkdu18 = .B;
   hhldkidu18 = .B;
   skdu18 = .B;
   skidu6 = .B;
   skd617 = .B;
   adltkd29 = .B;
   adltkd30 = .B;
   kidu5 = .B;
   kid = .B;
   kidu6 = .B;
   kid617 = .B;
   hhkid617 = .B;
   elder = .B;
   hhage = .B;
   hhsex = 'B';
   hhspan = 'B';
   hhrace = 'B';
   hhmove = .B;
   hhmar = 'B';
   hhgrad = .B;
   hhrel = .B;
   hhcit = 'B';
   hhinusyr = .B;
   spouse = 'B';
   nonrel = .B;
   coown = .B;
   nrldgr = .B;
   unrlkid = .B;
   unrlper = .B;
   parent = 'B';
   grandpa = 'B';
   grankid = 'B';
   refper = 'B';
   relfam = .B;
   relhhage = .B;
   reloth = .B;
   othrel = .B;
   nrel2 = 'B';
   norel3 = .B;
   nrel3 = 'B';
   scndry = .B;
   respdnt = 'B';
   hrmar = 'B';
   movedly = 'B';
   cntrace = 0;
   vother = 0;
   vother2 = 0;
   MOBILTYPE = 'B';
   QSELF = 'B';
   QINT = 'B';
   QDIV = 'B';
   QRENT = 'B';
   QSS = 'B';
   QSSI = 'B';
   QWELF = 'B';
   QRETIR = 'B';
   QWKCMP = 'B';
   QOTHER = 'B';
   QALIM = 'B';
   jsal = 'B';
 end;

 if status = '1' then
 do;

    if first.ctrlnm13 then
    do;
      zinc = 0;
      salinc = 0;
      zinc2 = 0;
      zadult = 0;
      hhldkid = 0;
      hhkdu18 = 0;
      hhldkidu18 = 0;
      skdu18 = 0;
      skidu6 = 0;
      skd617 = 0;
      adltkd29 = 0;
      adltkd30 = 0;
      kidu5 = 0;
      kid = 0;
      kidu6 = 0;
      kid617 = 0;
      hhkid617 = 0;
      elder = 0;
      hhage = 0;
      hhsex = ' ';
      hhspan = ' ';
      hhrace = ' ';
      hhmove = 0;
      hhmar = ' ';
      hhgrad = 0;
      hhrel = 0;
      hhcit = ' ';
      hhinusyr = 0;
      spouse = ' ';
      nonrel = 0;
      coown = 0;
      nrldgr = 0;
      unrlkid = 0;
      unrlper = 0;
      parent = ' ';
      grandpa = ' ';
      grankid = ' ';
      refper = ' ';
      relfam = 0;
      relhhage = 0;
      reloth = 0;
      othrel = 0;
      nrel2 = ' ';
      nrel3 = ' ';
      norel3 = 0;
      scndry = 0;
      respdnt = ' ';
      hrmar = ' ';
      movedly = ' ';
      hhblack = ' ';
      cntrace = 0;
      vother = 0;
      vother2 =0;
      MOBILTYPE = ' ';
      QSELF = ' ';
      QINT = ' ';
      QDIV = ' ';
      QRENT = ' ';
      QSS =  ' ';
      QSSI = ' ';
      QWELF = ' ';
      QRETIR = ' ';
      QWKCMP = ' ';
      QOTHER = ' ';
      QALIM = ' ';
      jsal = ' ';


    end;  /**  end of first.ctrlnm13 ck **/

 if hhmem='1' then
  do;

/* vother recode - total amount of non-wage income for family members */
  if age < 16 then pvother = .B;

  if (age >= 16 and rel < 30) then
     vother+pvother;

/* zinc recode - total income of reference person and household  */
/* members related to reference person                           */
/* salinc recode - salary income used to calculate poverty level */

    if (age >= 16 and rel < 30) then
     do;
      zinc = SUM(sal, pvother,zinc);
      salinc+sal;
     end;

/* zinc2 recode - total household income */
/* vother2 recode - total non-wage income for household  */

   if (age >= 16) then
    do;
      zinc2 = SUM(sal,pvother,zinc2);
      vother2+pvother;
    end;

/* zadult recodes  - COMP03 - number of adults over 18 */
    if (age >= 18) then
           zadult+1;

%countkids;
/* See below, Commented code was revised in include */

/* the following recodes all have to do with the children in the household  */
   if rel = 22 then
   do;

/* hhldkid recode -- used in setting the GEN3 recode  */
      hhldkid+1;

/* hhldkidu18 */
   if (0 <= age < 18) then
      hhldkidu18+1;


/* skdu18 recode - OWNC01 - counts the number of kids under the age of 18 */
      if ((mar = '6' and 13 < age < 18) or 0 <= age <= 13) then
      do;
           skdu18+1;

/* skidu6 recode - OWNC01 - counts the number of kids under the age of 6 */
        if 0 <= age < 6 then
          skidu6+1;

/* skd617 recode - OWNC01 - counts the number of kids between 6 and 17 */
        if 6 <= age <= 17 then
          skd617+1;
      end;

/* adltkd29 recode - COMP05 - counts the number of adult single children
   between the ages of 18 and 29 */
      if 18<=age<=29 and (mar ne '1' and mar ne '2') then
         adltkd29+1;

/* adltkd30 recode - COMP05 - counts the number of adult single children older
   than 29 */
      if age > 29 and (mar ne '1' and mar ne '2') then
         adltkd30+1;

   end;         /* end of rel = 22 check /*

/* kid recode - COMP03 and COMP04 - number of unmarried kids in a hhld */
   if ((13 < age < 18 and '3' <= mar <= '6') or ( 0 <= age <= 13)) then
   do;
      kid+1;

/* kidu5 - CHHS01  */
   if 0 <= age < 5 then
      kidu5+1;

/* kidu6 recode - COMP03 - counts the number of kids under the age of 6 */
      if 0 <= age < 6 then
         kidu6+1;

/* kid617 recode - COMP03 - counts the number of kids between 6 and 17 */
      if 6 <= age <= 17 then
         kid617+1;

   end;   /* end of check for single kids under 18 */

/*kid recode - CHHS01 recheck number of kids in a hhld single and those 
that might be married */

if 6 <= age <= 17 then 
   HHKID617 = HHKID617 + 1;

/* elder recode - AGE01 - number of elderly persons in a houshold */
   if (age ge 65) then
         elder+1;

/* hhage and hhsex recode - COMP01,COMP02,BOX02,MOVE04 - set recode to the age
   sex span race and move of the householder */
/* hhcit and hhinusyr also set for NATV01 and NATV02 */
   if (rel = 1 or rel = 2) then
   do;
          hhage = age;
          hhsex = sex;
          hhspan = span;
          hhrace = RACE;
          hhmove = move;
          hhmar = mar;
          hhgrad = grad;
          hhrel = rel;
          hhcit = citshp;
          hhinusyr=inusyr;
          hhblack = black;
   end;

/* spouse recode - COMP05 - whether a spouse is present */
   if rel = 20 then
      spouse = 'X';

/* nonrel recode - COMP05 - number of non-relatives */
/* coown recode - COMP05 - number of coowners */
/* nrldgr recode - COMP05 - non-relative lodger */
/* unrelkid recode - COMP05 - number of unrelated children */
/* unrelper recode - COMP05 - number of unrelated persons */
   if rel > 26 then
     do;
       nonrel+1;
       if ten = '1' then
          coown+1;
       else
       if lodsta = '1' then
          nrldgr+1;
       else
       if ten ne '1' and lodsta ne '1' then
         do;
           if 0 <= age <= 17 then
              unrlkid+1;
           else
              unrlper+1;
         end;
       else
         do;
           if 0 <= age <= 17 then
              unrlkid+1;
           else
              unrlper+1;
         end;
      end;

/* COMP05 - recodes used to determine three generations in residence */
   if rel = 24 then
   do;
      parent = 'X';
      if par gt 0 then
         grandpa = 'X';
   end;
   if rel = 23 then
      grankid = 'X';
   if rel = 1 or rel = 2 then
      refper = 'X';

/* relfam and relhhage recodes  - COMP05 */
   if famnum >= '2' and famtyp = '3' and famrel = '1' then
      do;
        relfam+1;
        relhhage = age;
      end;

/* reloth recode  - COMP05 */
   if famrel = '4' then
      reloth+1;

/* othrel recode  - COMP05*/
   if (rel > 22 and rel <= 26) or (rel=22 and age ge 18) then
      othrel+1;

/* nrel2 - COMP05 - 2 person hhld-not related */
   if curper = 2 and rel = 2 then
      nrel2 = 'X';

/* nrel3 - COMP05 - 3 to 8 person hhld-not related, this is the
   first part, the actual recode is set when the last control
   number for the group of persons is found */
   if 3 <= curper <= 8 and (rel = 2 or rel = 32 or rel = 34
      or rel = 36 or rel = 38) then
      norel3+1;

/* scndry - COMP05 - one or more secondary families */
   if famnum >= '2' and famtyp = '4' then
      scndry+1;

/* respdnt recode -- recode for household respondent who moved in the last
    year   --  and also the recode hrmar -- houshold respondents marital
    status  */
    if dline1 = pline and '1' <= mvg <= '4' and (datey = move or
       (datey - move = 1 and movm >= datem)) then
        do;
          respdnt = 'Y';
          hrmar = mar;
        end;

/* movedly recode -- recode for determining if houshldr moved in the last
    year   */
    if  (rel = 1 or rel = 2) and (datey = move or (datey - move = 1 and
       movm >= datem)) then
       movedly = 'X';


/**********************************************************************
** Beginning with N05 ...                                            **
** The include statement below runs the following macro for the      **
** New income variables for reference person and related individuals **
** (family level) for Receipt of ... income                          **
**********************************************************************/
%include "$PROG/inc_recpt_incl.sas"/source2;

%recptinc;

/************************************************************************
* Starting with N05 ...                                                 *
* The following include statement runs the two macros -                 *
************************************************************************/
%include "$PROG/mob_asst_incl.sas"/source2;

/* Recode the responses to the manufactured/mobile home questions into  */
%mobil;

/* The Eleven (11) assisted living services questions into two recodes  */
/*  in 2011 these questions were not asked - do not call macro */
/*  %asstlvg;  */

   end;  /* end of hhmem check */

 end;   /* end of status check */

if last.ctrlnm13 then
   do;
   if norel3 = curper then nrel3 = 'X';
   output rcds.perrcd;
   outcnt+1;
end;

if send then
do;
  put ctrlno ' = number of first control numbers processed';
  put ctllst ' = number of last control numbers processed';
  put incnt ' = number of obs input';
  put outcnt ' = number of obs output';
end;

/* end of recode program */
run;

/*  sect2_hh_inc.sas   */
/* ********************************************************************
*  Program: Sect2_hh_inc.sas
*   Author:  Diane Schwartz
*   Background: In 2011 the table recode program was split into sections
*      so that portions could be called to run in the PUF recodes or
*      in the general recodes.                                       
* ******************************************************************* */

/* *******************************************************************
*           S E C T I O N   T W O
*
*      now do the processing not involving the persons data set or
*      any of the other
*      rostered data sets!                                           
* ******************************************************************* 
*  Section Two -- main portion of recodes
*    phone  --  phone
*    kitchen -- complete kitchen
*    rooms -- number of rooms
*    confeet - confeetr -- rounded at end to be confeet -- condominium and
*                          cooperative fee
*    zcrowf - zcrowfr -- rounded at end to be zcrowf -- square feet per
*                        person
*    zcrowd - zcrowdr -- rounded at end to be zcrowd -- persons per room
*    zcrowb - zcrowbr -- rounded at end to be zcrowb -- persons per bedroom
*    tprice -- purchase price
*    PCTDWN -- Percent Down - added in 2007
*    zaq1 -- tally for upkeep in selected physical problems
*    zadeq -- tally to determine severe or moderated selected physical problems
*    maint - maintr -- rounded at end to be maint -- routine maintenance in
*                      last year
*    zsmhcn - zsmhcnr -- rounded at end to be zsmhcn -- monthly housing costs
*                        withour maintenance
*    zsmhcm - zsmhcmr -- rounded at end to be zsmchm -- monthly housing costs
*                        with maintenance
*    zsmhco - zsmhcor -- rounded at end to be zsmhco -- monthly housing costs
*                        excluding 2nd and subsequent mortgages and maintenance
*    zsmhcp - zsmhcpr -- rounded at end to be zsmhcp -- monthly housing costs
*                        as a percent of current income
*    zvi - zvir -- rounded at end to be zvi -- ratio of value to current income
*    amtomo - amtomor -- rounded at end to be amtomo -- average monthly costs
*                        for fuel oil
*    amtimo - amtimor -- rounded at end to be amtimo -- average monthly costs
*                        for property insurance
*    amtwmo - amtwmor -- rounded at end to be amtwmo -- average monthly costs
*                        for water
*    amttmo - amttmor -- rounded at end to be amttmo -- average monthly costs
*                        for trash
*    amtgmo - amtgmor -- rounded at end to be amtgmo -- average monthly costs
*                        for bottled gas
*    amtfmo - amtfmor -- rounded at end to be amtfmo -- average monthly costs
*                        for other fuel
*    lrntmo - lrntmor -- rounded at end to be lrntmo -- land rent fee paid
*    amtxp - amtxpr -- rounded at end to be amtxp -- average monthly costs
*                      for real estate taxes
*    amtxv - amtxvr -- rounded at end to be amtxv -- annual taxes paid per
*                      $1000 value
*    newc -- new construction
*    pubsub -- rent reductions
*    vehic -- number of vehicles in houshld
*
************************************************************************** */


/* ** this section creates a data set with the recodes included. ** */

data rcds.hhrcd (keep = &msrtkey phone kitchen zcrowfr zcrowdr zcrowbr
                          rooms confeetr zaq1  zadeq maintr zsmhcnr
                          zsmhcmr zsmhcor zsmhcpr zvir amtxpr
                          amtxvr tprice amtomor amtimor amtwmor amtgmor
                          amttmor amtfmor lrntmor newc pubsub
                          PCTDWNR ctrlnum lotacre VEHIC);

    length PHONE $1 KITCHEN $1 zcrowfr 3. zcrowdr 3. zcrowbr 3.
           ROOMS 3. confeetr 3. ZADEQ $1 zsmhcnr 4. zsmhcmr 4.
           zsmhcor 4. zsmhcpr 4. zvir 4. amtxpr 4. amtxvr 4.
           TPRICE 4. amtomor 3. amtimor 3. amtwmor 3. amtgmor 3. amttmor 3.
           amtfmor 3. lrntmor 3. NEWC $1 ZAQ1 3. maintr 4.
           PUBSUB $1 PCTDWNR 4. VEHIC 4.;

     merge houshld (in=hldin) end=send
           uhoushldr(keep= &msrtkey phonenum telhh telav ctrlnum)
           mortg
           rcds.perrcd;
      by &msrtkey;

    retain incnt 0;
    retain ctrlno 0;
    retain ctllst 0;
    retain outcnt 0;
    retain newcm 0;
    retain lotacre 0;
/*  for 2007, you must use the edits houshld data set to control the obs that
*   will get processed since the research dataset has both the nat and ms cases
*   combined */
if hldin then
do;
    incnt+1;
   phone = 'B';
   kitchen = 'B';
   rooms = .B;
   confeetr = .B;
   zcrowfr = .B;
   zcrowdr = .B;
   zcrowbr = .B;
   tprice = .B;
   PCTDWNR = .B;
   zaq1 = .B;
   zadeq = 'B';
   maintr = .B;
   zsmhcnr = .B;
   zsmhcmr = .B;
   zsmhcor = .B;
   zsmhcpr = .B;
   zvir = .B;
   amtomor = .B;
   amtimor = .B;
   amtwmor = .B;
   amttmor = .B;
   amtgmor = .B;
   amtfmor = .B;
   lrntmor = .B;
   amtxpr = .B;
   amtxvr = .B;
   newc = 'B';
   pubsub = 'B';

 /* Recode to convert lot into acres for use in table to calculate
      median correctly */

 lotacre=lot/44000;

 if status > '0' and status < '4' then
 do;

/* phone recode - */
   if (status = '1') then do;
      if (phonenum ne ' ' or telhh = '1' or telav = '1') then phone = '1';
      else if (telhh = '2' and telav = '2') then phone = '2';
      else phone = 'D';
   end;

   kitchen = ' ';
/* kitchen recode  - EQUP01 -complete kitchen facilities */

    if(((nunit2 = '1' or nunit2 = '2' or nunit2 = '4')
         and (sink = '1')
         and (refr = '1')
         and (cook = '1' or burner = '1' or oven = '1'))
       or
         ((nunit2 = '3')
         and (kexclu = '1')
         and (sink = '1')
         and (refr = '1')
         and (cook = '1' or burner = '1' or oven = '1'))) then
     do;
        kitchen = '1';
     end;
     else
     do;
        if ((sink = '2')
           or (refr = '2')
           or (cook = '2' and burner = '2' and oven = '2')
           or (nunit2 = '3' and kexclu = '2')) then
           do;
              kitchen = '2';
           end;
     end;

/* rooms recode - ROOM01 - number of rooms */
   rooms = 0;
   rooms = sum(bedrms,kitch,living,dining,famrm,recrm,dens,othfn);

/* confeet recode - condominium or homeowners assoc. or mobile home monthly
                    fees  - COOP02 and COST04  */

   if confeeq = . then confeet = .;
   if confeeq = 0 then confeet = 0;
   if confeeq >= 1 then
   do;
     confeetr = (confeeq * camf) / 12;
  end;

/* zcrowf recode - SQFT02 - units square feet per person */

if status = '1' and (nunit2 = '1' or nunit2 = '4') then
 do;
  if unitsf = .D or unitsf = .R or unitsf = . then
    zcrowf = .D;
  else
  do;
    zcrowfr = unitsf/curper;
  end;
 end;

/* zcrowd recode - PPRM01 - persons per rooms */

if status = '1' then
  do;
    zcrowdr = curper/rooms;
  end;

/* zcrowb recode - PPRM02 - persons per bedrooms */

if status = '1' then
do;
  if bedrms = 0 then
    zcrowbr = 0;
  else
    zcrowbr = curper/bedrms;
end;

end;      /* end of status check */

/* tprice recode  - PUPR01 - purchase price */
if 1 <= lprice <= 9999998 then
   tprice = lprice;
else
if 1 <= cprice <= 9999998 then
   tprice = cprice;
else
if lprice = .d or lprice = .r or lprice = . then
   tprice = .d;
else
if cprice = .d or cprice = .r or cprice = . then
   tprice = .d;

/* PCTDWNR recode - DWNPMT01 - Down Payment */
 if status = '1' and tenure = '1' and
    ((cushom ge '1' and cushom le '4') or preocc = '2') and
    (dwnpay ge '1' and dwnpay le '7') then
    do;
      if 1 <= tprice <= 9999998 and 1 <= dwndlr <= 9999998 then
         PCTDWNR = (dwndlr/tprice) * 100;
      else
      if dwndlr = .D or dwndlr = .R then
         PCTDWNR = .D;
      else
      if DWNDLR = . then
         PCTDWNR = .D;
      else
      if TPRICE = .D and
       (1 <= DWNDLR <= 9999998 or DWNDLR = .D or DWNDLR = .R or DWNDLR = .)
      then
         PCTDWNR = .D;
     end;
 if PCTDWNR > 100 then
    PCTDWNR = 100;


/* zadeq recode - PROB01 and BOX02B - adequacy of housing */
if status = '1' then
do;
  zadeq = '0';
  zaq1 = 0;

/* calculate zaq1 */
  if leak = '1' then
    zaq1+1;
  if ileak = '1' then
    zaq1+1;
  if holes = '1' then
    zaq1+1;
  if cracks = '1' then
    zaq1+1;
  if bigp = '1' then
    zaq1+1;
  if rats = '1' then
    zaq1+1;

/*  calculate zadeq */
  if (baths < 2 and (hotpip = '2'
   or tub = '2'
   or toilet = '2'
   or sharpf = '1'))
   or
     (freeze = '1'
   and (numcold >= '3' and numcold <= '8'))
   or
     (buye = '1')
   or
     (nowire = '2'
   and plugs = '2'
   and (numblow >= '3' and numblow <= '8'))
   or
     (zaq1 ge 5) then
   do;
      zadeq = '3';
   end;

   if zadeq ne '3' then
   do;
     if (numtlt >= '3' and numtlt <= '8') or
        (hequip = 7) or
        (zaq1 = 3 or zaq1 = 4) or
        (kitchen = '2') then
          zadeq = '2';
   end;

     if zadeq ne '3' and zadeq ne '2' then
          zadeq = '1';
end;  /* end of status check for zadeq universe */


/* recode for maint - MTNC01 - routine maintenance in last year */

if cstmnt >= 0 then
 do;
   maintr = cstmnt / 12;
 end;
else
   maintr = cstmnt;


/* zsmhcn - monthly housing costs recode (without maintenance)
          - COST01 and BOX07 */

if amtxre < 0 then amtxre = 0;
if mhotfe < 0 then mhotfe = 0;
if cstmnt < 0 then cstmnt = 0;

if (status = '1' and (tenure = '1' or tenure = '2')
   and prent ne 0) then
do;

  zsmhcnr = 0;

  if (1 <= amte <= 998) then zsmhcnr+amte;
  if (1 <= amtg <= 998) then zsmhcnr+amtg;
  if (amto >= 4) then zsmhcnr+(amto / 12);
  if (amtf >= 4) then zsmhcnr+(amtf / 12);
  if (amtt >= 4) then zsmhcnr+(amtt / 12);
  if (amtw >= 4) then zsmhcnr+(amtw / 12);

  if ((amtxq >= 0) and (taxpmt ne '1' and txpmt2 ne '1')) then
  do;
    zsmhcnr+((amtxq - amtxre) / 12);
  end;

  if (amti >= 1) and (inspmt ne '1' and inpmt2 ne '1') then
  do;
    zsmhcnr+(amti / 12);
  end;

  if (confeeq >= 1) then
     zsmhcnr+((confeeq*camf)/12);
  if ( 1 <= lrent <= 1996) then
     zsmhcnr+((lrent * flrent) / 12);
  if (mhotfe >= 1) then
     zsmhcnr+((mhotfe * fmhotf) / 12);
  if ((rent >= 2) and (prent < 1 or prent > 9997)) then
     zsmhcnr+((rent*frent)/12);
  if (prent >= 1 and frent >= 1) then
     zsmhcnr+((prent*frent) / 12);
  if (pmt >= 1) then
     zsmhcnr+pmt;
  if (pmt2 >= 1) then
     zsmhcnr+pmt2;
  if (amtm >= 1) then
     zsmhcnr = zsmhcnr-(amtm / 12);
  if (amtm2 >= 1) then
     zsmhcnr = zsmhcnr-(amtm2 / 12);
  if (pmiamt >= 1) then
     zsmhcnr = zsmhcnr - (pmiamt/12);
  if (pmamt2 >= 1) then
     zsmhcnr = zsmhcnr - (pmamt2/12);

end;  /* end of universe check for zsmhc */

/* zsmhcm - monthly housing costs recode (with maintenance) - COST12
   and COST13  */

if (status = '1' and tenure = '1')then
do;

  zsmhcmr = 0;

  if (1 <= amte <= 998) then zsmhcmr+amte;
  if (1 <= amtg <= 998) then zsmhcmr+amtg;
  if (amto >= 4) then zsmhcmr+(amto / 12);
  if (amtf >= 4) then zsmhcmr+(amtf / 12);
  if (amtt >= 4) then zsmhcmr+(amtt / 12);
  if (amtw >= 4) then zsmhcmr+(amtw / 12);

  if ((amtxq >= 0) and (taxpmt ne '1' and txpmt2 ne '1')) then
  do;
    zsmhcmr+((amtxq - amtxre) / 12);
  end;

  if (amti >= 1) and (inspmt ne '1' and inpmt2 ne '1') then
  do;
    zsmhcmr+(amti / 12);
  end;

  if (confeeq >= 1) then
     zsmhcmr+((confeeq*camf)/12);
  if ( 1 <= lrent <= 1996) then
     zsmhcmr+((lrent * flrent) / 12);
  if (mhotfe >= 1) then
     zsmhcmr+((mhotfe * fmhotf) / 12);
  if (pmt >= 1) then
     zsmhcmr+pmt;
  if (pmt2 >= 1) then
     zsmhcmr+pmt2;
  if (amtm >= 1) then
     zsmhcmr = zsmhcmr-(amtm / 12);
  if (amtm2 >= 1) then
     zsmhcmr = zsmhcmr-(amtm2 / 12);
  if (pmiamt >= 1) then
     zsmhcmr = zsmhcmr - (pmiamt/12);
  if (pmamt2 >= 1) then
     zsmhcmr = zsmhcmr - (pmamt2/12);
  if ((cstmnt / 12) >= 1) then
     zsmhcmr+(cstmnt / 12);

end;  /* end of universe check for zsmhcm */

/* zsmhco - monthly housing costs recode  - COST12 and COST13 */

if (status = '1' and tenure = '1') then
do;

  zsmhcor = 0;

  if (1 <= amte <= 998) then zsmhcor+amte;
  if (1 <= amtg <= 998) then zsmhcor+amtg;
  if (amto >= 4) then zsmhcor+(amto / 12);
  if (amtf >= 4) then zsmhcor+(amtf / 12);
  if (amtt >= 4) then zsmhcor+(amtt / 12);
  if (amtw >= 4) then zsmhcor+(amtw / 12);

  if ((amtxq >= 0) and (taxpmt ne '1' and txpmt2 ne '1')) then
  do;
    zsmhcor+((amtxq - amtxre) / 12);
  end;

  if (amti >= 1) and (inspmt ne '1' and inpmt2 ne '1') then
  do;
    zsmhcor+(amti / 12);
  end;

  if (confeeq >= 1) then
     zsmhcor+((confeeq*camf)/12);
  if ( 1 <= lrent <= 1996) then
     zsmhcor+((lrent * flrent) / 12);
  if (mhotfe >= 1) then
     zsmhcor+((mhotfe * fmhotf) / 12);
  if (pmt >= 1) then
     zsmhcor+pmt;
  if (amtm >= 1) then
     zsmhcor = zsmhcor-(amtm / 12);
  if (pmiamt >= 1) then
     zsmhcor = zsmhcor - (pmiamt/12);

end;  /* end of universe check for zsmhco */

/* recode for zsmhcp - monthly housing costs as a percent of income '
   - COST03 */

if status = '1' and (tenure = '1' or tenure = '2') and prent ne 0 then
do;
   zsmhcpr = 0;
   if zinc ge 1 then
    do;
     zsmhcpr = ((zsmhcnr * 12) / zinc) * 100;
    end;
   else if zinc < 1 and zincn ge 1 then
    do;
     zsmhcpr = ((zsmhcnr * 12) / zincn) * 100;
    end;
   else
    do;
     zsmhcpr = .X;
    end;


end;   /* end of universe check for zsmhcp */
/* recode for zvi - RATI01 - value as a percent of income */

 if (status = '1' and tenure = '1') then
 do;

   zvir = 0;
   if zinc ge 1 then
    do;
     zvir = value / zinc;
    end;
   else if zinc < 1 and zincn ge 1 then
    do;
     zvir = value / zincn;
    end;
   else
    do;
     zvir = .;
    end;

 end;   /* end of universe check for zvi */

/* recode for amtomo  - monthly cost for amto */
   if status = '1' and (buyo < '1' or buyo > '3')
     and billo ne '2' then
   do;
     amtomor = amto / 12;
   end;

/* recode for amtimo  - monthly cost for amti */
   if buyi = '1' then
   do;
     amtimor = amti / 12;
   end;

/* recodes for monthly costs for water (AMTWMO), trash (AMTTMO), bottled
   gas (AMTGMO), and other fuel (AMTFMO) - COST06 */
   if amtw ge 4 then
   do;
    amtwmor = amtw / 12;
   end;
   if amtt ge 4 and billwt ne '1' then
   do;
    amttmor = amtt / 12;
   end;
/*  amtg is already monthly, do not need to divide by 12 */
   if amtg ge 1 and gaspip = '2' and billtg ne '1' and billwg ne '1' then
   do;
     amtgmor = amtg;
   end;
   if amtf ge 4 and billtf ne '1' and billwf ne '1' then
   do;
    amtfmor = amtf / 12;
   end;

/* recode for monthy cost for land rent (LRNTMO) - COST04 */
  if (incs = '1' and lanpmt = '2') or (incs = '1' and mg = '2') then
  do;
    if flrent >= 0 and lrent >= 0 then
    do;
      lrntmor = (flrent * lrent) / 12;
    end;
    else
      lrntmor = 0;
  end;  /* universe check for lrntmo */

/* amtxp recode - monthly real estate taxes  - TAXS01 */

if((status = '1' and tenure = '1') or ((status = '2' or status = '3') and
  (vacancy = 3 or vacancy = 5))) then
  do;
    amtxpr = .;
    if ((status = '1' and tenure = '1') and txre ne '1' and amtxq = 0) then
       amtxpr = 0;
    else
       if((status = '1' and tenure = '1') and txre = '1' and
         ((amtxq - amtxre) <= 0)) then
           amtxpr = 0;
    else
       if((status = '2' or status = '3') and (vacancy = 3 or vacancy = 5)
         and amtxq <= 0) then
           amtxpr = 0;
    else
       if((status = '1' and tenure = '1') and txre ne '1' and amtxq >= 1)
         then
           amtxpr = amtxq / 12;
    else
       if((status = '1' and tenure = '1') and txre = '1' and
         ((amtxq - amtxre) >= 1)) then
           amtxpr = (amtxq - amtxre) / 12;
    else
       if((status = '2' or status = '3') and (vacancy = 3 or
          vacancy = 5) and amtxq >= 1) then
           amtxpr = amtxq / 12;



end;  /* end of universe check for amtx */

/* recode for amtxv - real estate taxes as a percent of value - TAXS02 */
if (status = '1' and tenure = '1') or ((status = '3' or status = '2') and
   (vacancy = 3 or vacancy = 5)) then
do;
    amtxvr = (amtxpr * 12) / (value / 1000);

end;

/* recode for (Total Units) Publication -- NEWC - New construction last
   4 years - BOX02 */

  newc = ' ';

  if '1' <= status <= '3' then
  do;
     if (datey - 3) <= built <= datey then
        newc = '1';
     else
        if built = (datey - 4) then
         do;
           if nunit2 = '4' then
            do;
             if ranuni (datey) <= .25 then
               newc = '1';
            end;
           else
             if mbuilt ge datem then
               newc = '1';
         end;
  end;  /* end of universe check for newc */


/* Recode for PUBSUB -- RENT02 -- Rent reductions  */

  if (status = '1' and (tenure = '2' or tenure = '3')) or
     ((status = '2' or status = '3') and (vacancy = 1 or
     vacancy = 2 or vacancy = 4)) then
  do;
      pubsub = ' ';
      if proj = '1' then
        pubsub = '6';
      else
        if (subrnt = '1' and proj ne '1') or vcher = '1' or
           apply = '1' then
           pubsub = '7';
      else
        if renew = '1' and proj ne '1' and subrnt ne '1' and
           vcher ne '1' and apply ne '1' then
           pubsub = '8';
      else
        if rcntrl = '1' then
           pubsub = '1';
      else
        if subrnt = '2' and renew = '2' and (rcntrl = 'D' or rcntrl = 'R' or
           rcntrl = ' ') then
           pubsub = '5';
      else
        if rcntrl = '2' and rntadj = '1' then
           pubsub = '2';
      else
        if rcntrl = '2' and rntadj = '2' then
           pubsub = '3';
      else
        if rcntrl = '2' and ((rntadj = 'D' or rntadj = 'R' or
           rntadj = ' ') or ((status = '2' or status = '3') and
           (vacancy = 1 or vacancy = 2 or vacancy = 4) and (rntadj = 'B'))) then
           pubsub = '4';
      else
        pubsub = '9';
  end;  /*  end of universe check for pubsub */

/* recode for VEHA01 - number of vehicles in the houshld */
if status = '1' then
    VEHIC = cars+trucks;


output;
outcnt+1;
end;  /* end of check for controlling the observations processed by checking
*  hldin, in the houshld dataset */

if send then
do;
  put incnt ' = number of view obs input';
  put outcnt ' = number of obs output';
end;
run;

/*  sect3_poor_inc.sas   */
/* **************************************************************** */
/*               S E C T I O N   T H R E E                          */
/*                                                                  */
/*  Calculate POOR by using a poverty look up table                 */
/* **************************************************************** */
/*  Section Three -- poverty lookup table
*    poor - poorr -- rounded at end to be poor -- household income as percent
*                    of poverty level
*    pvlkup -- not used in tables, only for verification -- key to access the
*              poverty lookup table
*    povlvl -- not used in tables, only for verification -- value in poverty
*              lookup table corresponding to lookup key
*/

data rcds.povrcd (keep= &msrtkey poorr pvlkup povlvl ctrlnum);

 length POORR 4. POVLVL 4.;

  set  rcds.perrcd;
       by &msrtkey;


   attrib pvlkup format=$5.
          agehh pers kids format=$1.
          imonth format=$2.;

     if status = '1' then
      do;
%calcpov;

/* retrieve poverty threshhold level from the previously created poverty
*  formats  */

      povlvl=input(put(pvlkup,$povfmt.),6.);

/* calculate the income as percent of poverty level */

      if povlvl ne 0 then
       do;
         if zinc2 ne 0 then
          do;
            poorr=(zinc2*1000)/(povlvl);
          end;
         else poorr=1;
       end;
      else
         put ' error in poverty lookup for ctrlnm13 = ' ctrlnm13;

/* check end ranges for the calculated poor variable */

      if poorr lt 1 then
        poorr=1;

   end;  /* status check */
   output;
run;

/*  sect4_incm05_inc.sas  */
/*  ***************************************************************** */
/*            S E C T I O N     F O U R                               */
/*                                                                    */
/*   This section calculates the inc3flg for the INCM05 stub          */
/* ****************************************************************** */

data rcds.inc3rcd (keep= &msrtkey salinc inc3flg zinc status curper
                  per20wg ctrlnum)
     rcds.inc3rcdp (keep= &msrtkey salinc per20wg sal age status curper
                  rel ctrlnum);

 merge rcds.perrcd (in=inc3)
       houshld (keep = &msrtkey status curper ctrlnum)
       person (in=pers);
          by &msrtkey;

 length INC3FLG $1;
 retain PER20WG 0;
 retain CNTR 0;

 if status = '1' then
 do;
   if first.ctrlnm13 then
     do;
      cntr = 0;
      inc3flg = ' ';
     end;
   if (age ge 16 and ((1<=rel<=2) or rel = 20 or (22 <= rel <=26))) then
    do;
      per20wg = salinc * .2;
      if (sal gt 0 and sal gt per20wg) then
        cntr+1;
    end;
 end;  /* end of status check */

   if last.ctrlnm13 then
     do;
       if cntr > 1 then
         inc3flg = 'X';
       else
         inc3flg = ' ';
       output rcds.inc3rcdp;
       output rcds.inc3rcd;
     end;
   else
      output rcds.inc3rcdp;

run;

/*  sect5_mortg_pt1_inc.sas   */
/* **************************************************************************
*             S E C T I O N   F I V E
*
*    The following code recodes the mortgage variables
*          UNPBALT  -  outstanding principle 
*          CLPEVA -  current loan as percent of value
*          CRINTR -  current interest rate
*          PMTT - Monthly Mortgage Payment
*          PTCHAMR - Monthly Payment Change of Primary Mortgage
*
**************************************************************************** */


data rcds.mtgrcd (keep = &msrtkey unpbalt clpevar crintrr ctrlnum 
                         pmtt );

length  UNPBALT 4. CLPEVAR 4. CRINTRR 4. IINTF 3. IINTF2 3. PMTT 4.
        ;

 merge houshld
       mortg;
    by &msrtkey;

  unpbalt = .B;
  clpevar = .B;
  crintrr = .B;
  pmtt = .B;

     if status = '1' then
      do;

/* recode for UNPBALT */
   unpbalt = .;

   if mcnt ge 1 then do;
     UNPBALT = sum(UNPBAL,UNPBAL2,UNPBAL3,UNPBAL4,.);
   end;


/* recode for clpeva */

if tenure = 1 then 
do;

  clpevar = 1;
if 1 <= mcnt <= 7 then
do;
  clpevar = .;
  if UNPBALT ne . then
  do;
     clpevar = (UNPBALT / value) * 100;
  end;
  if value lt 10000 then
     clpevar =  .;
end;  /* end of mcnt check */
end;  /* end of tenure check */

/* recode for crintr */

if 1 <= mcnt <= 7 then
do;
  crintrr = .;
  iintf=input(intf,3.);
  if mcnt = 1 then
  do;
    crintrr = intw + (iintf/8);
  end;
  if mcnt => 2 then
  do;
    iintf2=input(intf2,3.);
    if (unpbal ne .D and unpbal ne .R and unpbal ne .) and
       (unpbal2 ne .D and unpbal2 ne .R and unpbal2 ne .) then
    do;
       crintrr = (((intw+(iintf/8))*unpbal)+((intw2+(iintf2/8))*unpbal2))/
                           (unpbal+unpbal2);
    end;
    if unpbal in (.D,.R,.B) or unpbal2 in (.D,.R,.B) then
      crintrr = .D;
  end;
end;

/*  recode for PMTT - MORT38 - Monthly Mortgage Payment */

pmtt = .;
if mcnt ge 1 then
   PMTT = sum(PMT,PMT2,PMT3,PMT4,.);

end;   /* end of status check */

output;
run;

/*  sect5a_mortg_inc.sas       */
/* **************************************************************************
*             S E C T I O N   F I V E
*
*    The following code recodes the mortgage variables
*          PMTIN  -  monthly mortgage payment as percent of income
*
**************************************************************************** */


data rcds.pmtinrcd (keep = &msrtkey PMTINR);

length  PMTINR 4. ;

 merge houshld (keep=&msrtkey status tenure zincn)
       rcds.mtgrcd (keep=&msrtkey PMTT)
       rcds.perrcd (keep=&msrtkey zinc);
    by &msrtkey;

  PMTINR = .B;

     if status = '1' then
      do;

      PMTINR = 0;
      if ZINC ge 1 then 
         PMTINR=((PMTT*12/ZINC) * 100);
      else 
      if ZINC lt 1 and ZINCN ge 1 then
         PMTINR=((PMTT*12/ZINCN) * 100);
      else
         PMTINR = .;
      

     end;   /* end of status check */

output;
run;

/*  sect6_mortg_pt2_inc.sas    */
/* ******************************************************************
*                S E C T I O N     S I X
*
*     Second part of mortgage recodes including
*        RYMORT -- remaining years mortgaged
*        PRIN01 -- monthly payment for principle and interest
*        PAYPM1 -- payment plan for first mortgage
*        PAYPM2 -- payment plan for second mortgage
*        the next 4 variables were added for n2001 processing -- DS
*        CRDLMT -- total home-equity line-of-credit limit
*        CRDLNS -- total outstanding line-of-credit loans
*        CRDINT -- current line-of-credit interest rate
*        CRDPMT -- line-of-credit monthly payment
******************************************************************** */


data rcds.mtg2rcd (keep = &msrtkey rymort prin01r paypm1 paypm2
                         crdlmt crdlns crdint crdpmt ctrlnum pmorig
                         rymort1 rymort2);

length RYMORT 4. PAYPM1 $1 PAYPM2 $1 PRIN01R 4. iheinf1 3. iheinf2 3.
       CRDLMT 4. CRDLNS 4. CRDINT 3. CRDPMT 4. PMORIG 3.;

 merge houshld
       mortg;
    by &msrtkey;


prin01r = .;

if status = '1' then
do;
/* recode for PMORIG - MORT08 - Year Primary Mortgage originated */
 if matbuy = '1' then
    pmorig = whnget;
 if matbuy = '2' then
    pmorig = yrmor;
/* recode for rymort - MORT10 - remaining years mortgaged */
 if 1 <= mcnt <= 7 and canvar ne '1' and canvr2 ne '1' then
 do;
  if mcnt = 1 then
  do;
    if matbuy = '1' then
      rymort = term - (datey-whnget);
    if matbuy = '2' then
      rymort = term - (datey-yrmor);
  end;

  if 2 <= mcnt <= 7 then
  do;
    if matbuy = '1' then
       rymort1 = term - (datey - whnget);
    if matbuy = '2' then
       rymort1 = term - (datey-yrmor);
    if matby2 = '1' then
       rymort2 = term2 - (datey - whnget);
    if matby2 = '2' then
       rymort2 = term2 - (datey - yrmor2);
    if rymort1 >= rymort2 then
       rymort = rymort1;
    if rymort1 < rymort2 then
       rymort = rymort2;
    if rymort < 0 then
       rymort = 0;
  end;
 end;

/* recode for prin01 -- monthly payment for principal and interest */

  if tenure = '1' and 1 <= mcnt <= 7 then
  do;
    /* prin01 = 0;  */
    prin01r = 0;
    if mcnt = 1 then prin01r+pmt;
    if mcnt = 2 then prin01r+pmt+pmt2;
    if (taxpmt = '1' or txpmt2 = '1') and 1 <= amtxq <= 99998 then
        prin01r = prin01r - (amtxq / 12);
    if (inspmt = '1' or inpmt2 = '1') and 1 <= amti <= 9998 then
        prin01r = prin01r - (amti / 12);
    if othpmt = '1' and 1 <= amtm <= 9999998 then
        prin01r = prin01r - (amtm / 12);
    if otpmt2 = '1' and 1 <= amtm2 <= 9999998 then
        prin01r = prin01r - (amtm2 / 12);
    if pmipmt = '1' and 1 <= pmiamt <= 999998 then
       prin01r = prin01r - (pmiamt/12);
    if pmpmt2 = '1' and 1 <= pmamt2 <= 999998 then
       prin01r = prin01r - (pmamt2/12);
    if prin01r < 0 then
       prin01r = 0;
  end;

/* recode for paypm1 - 15 - payment plan of primary mortgage */

  if mcnt < 1 or mcnt > 7 then
     paypm1 = 'B';
  if 1 <= mcnt <= 7 then
  do;
     paypm1 = 'D';
     count = 0;

     if arm = '1' then count+1;
     if gpmwp = '1' then count+1;
     if bloon = '1' then count+1;
     if varm = '1' then count+1;
     if timbom = '1' then count+1;
     if io = '1' then count+1;
     if canvar = '1' then count+1;

     if count ge 2 then
       paypm1 = '7';
     else
       if arm = '1' then 
         paypm1 = '2';
       else
       if canvar = '1' then
         paypm1 = '3';
       else
       if bloon = '1' then
         paypm1 = '4';
       else
       if gpmwp = '1' then 
         paypm1 = '5';
       else
       if varm = '1' then
         paypm1 = '6';
       else
       if timbom = '1' then
         paypm1 = '8';
       else
       if io = '1' then
         paypm1 = '9';
       else
       if vary = '1' or fixed = '1' then
         paypm1 = '1';
       else
       if vary = 'D' or vary = 'R' or vary = ' ' then
         paypm1 = 'D';
       else
       if arm in ('D','R','B') and bloon in ('D','R','B') and
          gpmwp in ('D','R','B') and varm in ('D','R','B') and
          timbom in ('D','R','B') and io in ('D','R','B') and
          fixed in ('D','R','B') then
         paypm1 = 'D';
  end; /* mcnt  = 1 - 4 check for paypm1 recode */
  else
     paypm1 = 'B';

/* paypm2 recode - mort16 - payment plan of secondary mortgage */
  if mcnt < 2 or mcnt > 7 then
     paypm2 = 'B';
  if 2 <= mcnt <= 7 then
  do;
     paypm2 = 'D';
     count2 = 0;
     if arm2 = '1' then count2+1;
     if gpmwp2 = '1' then count2+1;
     if bloon2 = '1' then count2+1;
     if varm2 = '1' then count2+1;
     if timbom2 = '1' then  count2+1;
     if io2 = '1' then count2+1;
     if canvr2 = '1' then count2+1;

     if count2 ge 2 then
        paypm2 = '7';
     else
      if arm2 = '1' then 
         paypm2 = 2;
      else
      if canvr2 = '1' then
         paypm2 = '3';
      else
      if bloon2 = '1' then
         paypm2 = '4';
      else
      if gpmwp2 = '1' then
         paypm2  = '5';
      else
      if varm2 = '1' then
         paypm2 = '6';
      else
      if timbom2 = '1'  then
         paypm2 = '8';
      else
      if io2 = '1' then 
         paypm2 = '9';
      else
      if vary2 = '1' or fixed2 = '1' then 
         paypm2 = '1';
      else
      if (vary2 = 'D' or vary2 = 'R' or vary2 = ' ') then
         paypm2 = 'D';
      else
      if arm2 in ('D','R','B') and bloon2 in ('D','R','B') and
         gpmwp2 in ('D','R','B') and varm2 in ('D','R','B') and
         timbom2 in ('D','R','B') and io2 in ('D','R','B') and
         fixed2 in ('D','R','B') then
         paypm2 = 'D';
  end;  /* end of mcnt = 2-4 check for paypm2 */
  else
     paypm2 = 'B';

/* CRDLMT recode -- total home-equity line-of-credit limit */
  crdlmt = .B;

  if helc = '1' then
  do;
     if (helcn = 1 or helcn = .D or helcn = .R) then
     do;
        if (1 <= hecr1 <= 9999998) then
            crdlmt = hecr1;
        else
            crdlmt = .D;
     end;
     else
     if helcn = 2 then
     do;
        if (1 <= hecr1 <= 9999998) and (1 <= hecr2 <= 9999998) then
           crdlmt = hecr1+hecr2;
        else
           crdlmt = .D;
     end;
    else
        crdlmt = .D;
    end;       /* end of helc check for crdlmt recode */


/* CRDLNS recode -- total outstanding line-of-credit loans */
    crdlns = .B;

    if hebal1 = '1' then
    do;
       if (helcn = 1 or helcn = .D or helcn = .R) or
          (helcn = 2 and hebal2 = '2') then
       do;
          if (1 <= hebam1 <= 9999998) then
             crdlns = hebam1;
          else
             crdlns = .D;
       end;
       else
       if (helcn = 2 and hebal2 = '1') then
       do;
          if (1 <= hebam1 <= 9999998) and (1 <= hebam2 <= 9999998)then
             crdlns = hebam1+hebam2;
          else
             crdlns = .D;
       end;
   end;    /*  end of hebal1 check for crdlns */

/* CRDINT recode -- current line-of-credit interest rate */
  crdint = .B;

  if hebal1 = '1' then
  do;
    if (helcn = 1 or helcn = .D or helcn = .R) or
       ((2 <= helcn <= 10 ) and hebal2 = '2') then
    do;
      iheinf1=input(heinf1,3.);
      if (1 <= hebam1 <= 9999998) and (1 <= heinw1 <= 25) and
         (0 <= iheinf1 <= 7) then
         crdint = heinw1 + (iheinf1 / 8);
      else
         crdint = .D;
    end;
    else
    if ((2 <= helcn <= 10) and hebal2 = '1') then
    do;
       iheinf1=input(heinf1,3.);
       iheinf2=input(heinf2,3.);
       if (1 <= hebam1 <= 9999998) and (1 <= heinw1 <= 25) and
          (0 <= iheinf1 <= 7) and (1 <= hebam2 <= 9999998) and
          (1 <= heinw2 <= 25) and (0 <= iheinf2 <= 7) then
          crdint = (((heinw1 + (iheinf1/8))*hebam1) +
                    ((heinw2 + (iheinf2/8))*hebam2)) /
                       (hebam1 + hebam2);
       else
          crdint = .D;
    end;
    else
       crdint = .D;
  end;  /*  end of hebal1 check for crdint recode */

/* CRDPMT recode -- line-of-credit monthly payment */
  crdpmt = .B;

  if hebal1 = '1' then
  do;
    if (helcn = 1 or helcn = .D or helcn = .R) or
       (helcn = 2 and hebal2 = '2') then
    do;
      if (1 <= hepmt1 <= 9998) then
         crdpmt = hepmt1;
      else
         crdpmt = .D;
    end;
    else
    if (helcn = 2 and hebal2 = '1') then
    do;
      if (1 <= hepmt1 <= 9998) and (1<= hepmt2 <= 9998) then
         crdpmt = hepmt1 + hepmt2;
      else
         crdpmt = .D;
    end;
  else
    crdpmt = .D;
  end;     /*  end of hebal1 check for crdpmt recode */

end;  /* end of status check */
run;

/*  sect7_move02_inc.sas   */
/* *********************************************************************
*               S E C T I O N    S E V E N
*
*  recent movers recode for table stub MOVE02
*********************************************************************** */

data mrgd1;
   merge houshld (keep= &msrtkey status datem datey curper almv ctrlnum dline1)
         person (keep= &msrtkey move movm pline rel mvg ctrlnum);
      by &msrtkey;
run;

proc sort data=mrgd1 out=mrgd2;
   by ctrlnm13 mvg;
run;
proc sort data=rmov;
   by ctrlnm13;
run;

data rcds.mv2rcd (keep = &msrtkey AMVONE MV1HH MV2HH AMVTWO SMV3 MV3HH
                        MV4NR LASTYR XHEADRES XUNITRES XCOSTRES XPERRES
                        XINUSRES XTENRES XRELRES RMVUNVRES CTRLNUM)
     rcds.mv2rec;

length AMVONE 4. AMVTWO 4. SMV3 4. MV1HH $1 MV2HH $1 MV3HH $1
       MV4NR $1 XHEADRES $1 XUNITRES $1 XCOSTRES $1 XPERRES 3.
       XINUSRES $1 XTENRES $1 XRELRES $1 RMVUNVRES $1 LASTYR 3.;

    merge mrgd2
          rmov (rename=(movgrp=mvg)) ;
    by ctrlnm13 mvg;

   retain lastyr 0;
   retain lastyrfl ' ';
   retain zeros 0;
   retain ones 0;
   retain twos 0;
   retain xheadres ' ';
   retain xhead1 ' ';
   retain lstmvg ' ';
   retain xunitres ' ';
   retain xcostres ' ';
   retain xperres .;
   retain xinusres ' ';
   retain xtenres ' ';
   retain xrelres ' ';
   retain rmvunvres ' ';
   retain totmov 0;

   if first.ctrlnm13 then
   do;
     lastyr = 0;
     amvtwo = 0;
     smv3 = 0;
     amvone = 0;
     lastyrfl = ' ';
     ones = 0;
     twos = 0;
     zeros = 0;
     xheadres = ' ';
     xhead1 = ' ';
     lstmvg = ' ';
     xunitres = ' ';
     xcostres = ' ';
     xperres = 0;
     xinusres = ' ';
     xtenres = ' ';
     xrelres = ' ';
     rmvunvres = ' ';
     totmov = 0;
   end;

   if status = '1' then
   do;

     if (move = datey) or ((move = datey-1) and (datem <= movm)) then
     do;
       lastyr+1;
       lastyrfl = 'X';
       if ('1' <= mvg <= '4' and lstmvg ne mvg) then
         totmov+1;
     end;  /*  with in last year check */

     if dline1 = pline then
     do;
        if xinus ne '1'  and
          ((move = datey) or ((move = datey-1) and (datem <= movm))) then
           rmvunvres = 'X';
     end;


     if mvg >= '1' and mvg <= '4' then
     do;
        if (move = datey) or ((move = datey - 1) and (datem <= movm)) then
        do;
         if lstmvg ne mvg then
         do;
           if (xhead = ' ' or xhead = 'D' or
                xhead = 'R' or xhead = 'B') then zeros+1;
           if xhead = '1' then ones+1;
           if xhead = '2' then twos+1;
           if mvg = '1' then xhead1 = xhead;
           lstmvg = mvg;
         end;

        if dline1 = pline then
          do;
          xheadres = xhead;
          xunitres = xunit;
          xcostres = xcost;
          xperres = xper;
          xinusres = xinus;
          xrelres = xrel;
          xtenres = xten;
        end;
       end;

     end;  /* mvg check */

     if last.ctrlnm13 and lastyrfl = 'X' then
     do;
        if almv = '1' and lastyr = curper then
        do;
          amvone+1;
          if xhead1 = '2' then
           mv1hh = '0';
          if xhead1 = '1' then
           mv1hh = '1';
          if xhead1 = 'D' or xhead1 = 'R' or xhead1 = ' '
             or xhead1 = 'B' then
           mv1hh = 'N';
        end;
        else
        if almv = '2' and lastyr = curper then
        do;
          amvtwo+1;
          if twos = totmov then
           mv2hh = '0';
          else
          if ones = 1 and (twos = totmov-1) then
           mv2hh = '1';
          else
          if ones >= 2 then
           mv2hh = '2';
          else
          if zeros ge 1 then
           mv2hh = 'N';
        end;
        else
        if lastyr lt curper then
        do;
          smv3+1;
          if twos = totmov then
            mv3hh = '0';
          else
          if ones ge 1 then
            mv3hh = '1';
          else
          if zeros ge 1 then
            mv3hh = 'N';
        end;
        else
        if lastyr = curper and (almv = ' ' or almv = 'D' or
                almv = 'R' or almv = 'B')then
          mv4nr = 'N';

     end;  /* end of processing last ctrlnm13 and mvg */

   end;    /* end of status = 1 check */

   output rcds.mv2rec;
   if last.ctrlnm13 then output rcds.mv2rcd;

run;

proc sort data=rcds.mv2rcd;
   by &msrtkey;
run;

proc sort data=rcds.mv2rec;
   by &msrtkey;
run;

proc sort data=rmov;
   by &msrtkey;
run;

/*   sect9_lodg_inc.sas   */
/* ******************************************************************** */
/*        S E C T I O N   N I N E
*
*  Recode for lodg and ldgrs:
*
*  Outputs one observation per household showing rent paid by
*    nonrelative lodgers
*    (Lodgers not related to the head of household and
*    not having parents or spouse that are co-owners or co-renters)    */
/* ******************************************************************** */

data rcdp.rnt01rcd (keep = &msrtkey lodgr lodgrs sgrndpa sparnt persct ctrlnum);

  length LODGR 4. LODGRS 4. SGRNDPA $1 SPARNT $1 SPRNTLN 3.;
  retain lodgr 0;
  retain lodgrs 0;


  ARRAY pers_(30)       ;
  ARRAY ldsta_(30) $1.  ;
  ARRAY rel_(30)        ;
  ARRAY par_(30)        ;
  ARRAY rntdu_(30)      ;
  ARRAY ldrnt_(30)      ;
  ARRAY ten_(30)   $1   ;
  ARRAY age_(30)        ;
  ARRAY spos_(30)       ;
  ARRAY ldgnr_(30) $1   ;

/* ** Read person observations for one household group and
      load person data into arrays ****** */
/*  ****************************************************************
*   ***** must use the person variable to load arrays, if you use the
*   ***** persct variable the line numbers may not match due to the 
*   ***** hhmem = 2 cases being dropped from the file, if you use the
*   ***** person variable, it accounts for the lines that were dropped
*   ****************************************************************** */
persct = 0;
DO UNTIL (LAST.&msrtkey);

   persct+1;
   merge houshld (keep = &msrtkey status ctrlnum)
         person;
     by &msrtkey;

   if status = '1' then
   do;
     pers_(person) = person;
     ldsta_(person) = lodsta;
     rel_(person) = rel;
     age_(person) = age;
     par_(person) = par;
     rntdu_(person) = rntdue;
     ldrnt_(person) = lodrnt;
     spos_(person) = spos;
     ten_(person) = ten;
   end;

END;  /* DO UNTIL loop */

/* Using the arrays created above, calculate only once arrays have been
   filled */
if last.&msrtkey then
do;
  persctln=person;

/* For each houshold, calculate and output household rent
     paid by nonrelatives */

if status = '1' then
do;

lodgrs = 0;
lodgr = 0;

DO i = 1 to persctln;


/*  check for gen3 - COMP05 - spouse parent and spouse grandparent check */
    if rel_(i) = 1 then
       if spos_(i) > 0 then
          if par_(spos_(i)) > 0 then
          do;
            sparnt = 'X';
            sparntln = par_(spos_(i));
            if par_(sparntln) > 0 then
            do;
               sgrndpa = 'X';
               put 'spouses grandparents -- ctrlnm13 = ' ctrlnm13;
            end;
          end;  /* parent of spouse check */

/*  this is the rent paid by lodger recode that has been rewritten for n99. */

if ldsta_(i) eq '1' then lodgrs=lodgrs +1;
if ldsta_(i) eq '1' and (rntdu_(i) ge 1 and rntdu_(i) le 52)
     and ldrnt_(i) ge 1 then
   lodgr=lodgr + (ldrnt_(i) * rntdu_(i)) / 12;
/* if (i eq persct) then
   do;
     if lodgr le 0 then
        lodgr = .D ;
   end;  */
END;   /* DO loop */

if lodgrs le 0 then 
   lodgr = .B;
if lodgr = 0 then
   lodgr = .D;

end;  /* status check */


OUTPUT;

/* ** Clear arrays before next household group is read and processed ** */
lodgrs = .B;
lodgr = .B;
sparnt = ' ';
sgrndpa = ' ';
sprntln = .B;
do i = 1 to 30;
  pers_(i)  = 0  ;
  ldsta_(i) = ' ';
  rel_(i) = 0    ;
  par_(i)   = 0  ;
  rntdu_(i)  = 0 ;
  ten_(i) = ' '  ;
  ldrnt_(i) = 0  ;
  age_(i) = 0    ;
  spos_(i) = 0   ;
  ldgnr_(i) = ' ';
end;
persctln=.;
end;  /* last.msrtkey check */
run;

/* **** The following 2 programs are run thru the script and then the datasets are merged with the data created in the table recode process. *******/
/**************************************************************************
*   This program create the recodes for Households with Disabled
*    Persons.
*
*   Program:      db_recodes.sas - Disabilities Recode
*   Programmed by: Ruth Coffie
*   Date:     July  29, 2009
*   Macros - written by Matthew Streeter (HHES).
*
*  There are six person level or "P" disability variables.  They are
*  PHEAR, PSEE, PMEMRY, PWALK, PCARE, and PERRND. The recodes create a
*  household disability variable (HDSB) and six household level, or
*  "H", variables.  These "H" variables correspond to the "P" variables.
*  They are HHEAR, HSEE, HMEMRY, HWALK, HCARE, and HERRND.
*  If at least one member of a household has a value of '1' for a P variable,
*  then that household's H variable will be set to 1.
*  For example, HHEAR is '1' if any person in the household has a value
*  of '1' for PHEAR.  If all of the values for the P variables in a
*  household are '2', then the H variable for that disability will be '2'.
*
*  If all of the P variables in a household are 'D' or 'R', then
*   the corresponding H variable will be set to 'D' or 'R' respectively
*  (because of the way the questions and bridge are set up, if one
*  person variable is 'D' or 'R' in a household, they all will be for eligible
*  members of that household).  There will also be a household level
*  general disability variable, HDSB.  This variable will be set to '1'
*  if any of the H variables are '1'.  It will be set to '2' if all of the H
*  variables are '2' (excluding 'B' values).  HDSB will be set to 'N' if
*  none of the H variables are '1', but at least one H variable is 'D'
*   or 'R'.
*
*/

libname inp "$INPDR";
libname inp2 "$INPDR2";
libname rcd "$RECDS";

%global msrtkey;
%let msrtkey=%sysget (MSRTKEY);

/* This is the macro for checking if an person is positive
for a particular disability.  If it is, then the household level
variable is switched to '1' and the counter for with disability
is increased */

%MACRO discheck(hdis=,pdis=);
        if &pdis eq '1' then
        do;
            &hdis = '1';
            withdis+1;
        end;
%MEND discheck;

/* This is a macro for checking if a person has 'D' or 'R' value.
Because of the way the data was asked and bridged, when one person in
the household has a 'D' or an 'R', then the whole household has it.
For this reason, the household level variable is set to 'D' or 'R'
when the person value is 'D' or 'R' respectively, only if the
household level is not already set to 1. */

%MACRO drcheck(hdis=,pdis=);
        if (&pdis eq 'D' or &pdis eq 'R' or &pdis eq ' ')
           and &hdis ne '1'
        then
           do;
              &hdis = &pdis;
              dkrfct+1;
        end;
%MEND drcheck;

proc sort data=inp.houshld out=hh;
by &msrtkey;
run;

proc sort data=inp2.person out=per;
by &msrtkey person;
run;

data dbdata;
merge per (in = pers)  hh (in = house);
by &msrtkey;
if house then output;
run;

data rcd.dbrcd (keep= &msrtkey hhear hsee hmemry hwalk hcare
              herrnd hdsb withdis dkrfct);
        set dbdata;
        by &msrtkey;

        retain HHEAR ' ';
        retain HSEE ' ';
        retain HMEMRY ' ';
        retain HWALK ' ';
        retain HCARE ' ';
        retain HERRND ' ';
        retain HDSB ' ';
        retain WITHDIS 0;
        retain DKRFCT 0;

        if STATUS eq 1 then
        do;
              /* The counters and household level variables are reset for
                each first unique control number. */

                if first.&msrtkey then
                do;
                    HHEAR = '2';
                    HSEE = '2';
                    HMEMRY = '2';
                    HWALK = '2';
                    HCARE = '2';
                    HERRND = '2';
                    HDSB = '2';
                    WITHDIS = 0;
                    DKRFCT = 0;
                end;

             /* Passing disability data into the macro checking for people
                   with disabilities. */

                %discheck(hdis=HHEAR,pdis=PHEAR)
                %discheck(hdis=HSEE,pdis=PSEE)
                %discheck(hdis=HMEMRY,pdis=PMEMRY)
                %discheck(hdis=HWALK,pdis=PWALK)
                %discheck(hdis=HCARE,pdis=PCARE)
                %discheck(hdis=HERRND,pdis=PERRND)

             /* Blanked values of household members not of eligible age
                  could affect the following check.  Therefore, the
                  drcheck should be moved outside of the if
                  last.&msrtkey statement. */

                /* The disability data is passed through the 'D' and 'R' macro
                check only for the last member of each household, since they
                will be the same as the other members if they are
                a 'D' or 'R'. */

                if last.&msrtkey then
                do;
                    %drcheck(hdis=HHEAR,pdis=PHEAR)
                    %drcheck(hdis=HSEE,pdis=PSEE)
                    %drcheck(hdis=HMEMRY,pdis=PMEMRY)
                    %drcheck(hdis=HWALK,pdis=PWALK)
                    %drcheck(hdis=HCARE,pdis=PCARE)
                    %drcheck(hdis=HERRND,pdis=PERRND)

                    if withdis gt 0 then /* Checks counter if household
                                         has hit positive DSB check. */
                        HDSB = '1'; /* Sets household level disabled
                                    flag to positive. */
                    else
                    if dkrfct gt 0 then /* Checks 'D' and 'R' counter
                                     for without disability universe.*/
                       HDSB = 'N'; /* Sets 'N' value for those with at
                             least one not reported value but no
                             definite disabilities. */
                end;

   end;   /* end status check */
   else
   do;
                if first.&msrtkey then
                do;
                    HHEAR = 'B';
                    HSEE = 'B';
                    HMEMRY = 'B';
                    HWALK = 'B';
                    HCARE = 'B';
                    HERRND = 'B';
                    HDSB = 'B';
                end;
    end;

        if last.&msrtkey then output rcd.dbrcd; /* This will output
                      one observation per household after the HDSB is set. */
run;


/**************************************************************************
*   This program create the recodes for Households with Home Accessibility
*    Persons.
*
*   Program:      hmd_recodes.sas - Home Accessibilities Recode
*   Programmed by: Ruth Coffie - Diane Schwartz
*   Date:    March  6, 2011
*   Macros - written by Matthew Streeter (HHES).
*
*  This program is based on the Disability program, just updated to run the
*  HMD variables.
*
*  There are twelve person level or "P" disability variables.  They are
*  PKCAB, POCAB, PSTOVE, PCTRUSE, PGETBR, PSINK, PFAUCET, PBTUB, PWSHWR,
*  PSTOOP, PREACH and PGRASP. The recodes create
*  twelve household level variables, or
*  "H", variables.  These "H" variables correspond to the "P" variables.
*  They are HKCAB, HCAB, HSTOVE, HCTRUSE, HGETBR, HSINK, HFAUCET,
*  HBTUB, HWSHWR, HSTOOP, HREACH and HGRASP.
*  If at least one member of a household has a value of '1' for a P variable,
*  then that household's H variable will be set to 1.
*  For example, HHEAR is '1' if any person in the household has a value
*  of '1' for PHEAR.  If all of the values for the P variables in a
*  household are '2', then the H variable for that disability will be '2'.
*
*  If all of the P variables in a household are 'D' or 'R', then
*   the corresponding H variable will be set to 'D' or 'R' respectively
*  (because of the way the questions and bridge are set up, if one
*  person variable is 'D' or 'R' in a household, they all will be for eligible
*  members of that household).
*
*/

libname inpdr "$INPDR";
libname inpdr2 "$INPDR2";
libname rcd "$RECDS";

%global msrtkey;
%let msrtkey=%sysget (MSRTKEY);

/* This is the macro for checking if an person is positive
for a particular disability.  If it is, then the household level
variable is switched to '1' and the counter for with disability
is increased */

%MACRO difcheck(hdiff=,pdiff=);
        if &pdiff eq '1' then
        do;
            &hdiff = '1';
            withdiff+1;
        end;
%MEND difcheck;

/* This is a macro for checking if a person has 'D' or 'R' value.
Because of the way the data was asked and bridged, when one person in
the household has a 'D' or an 'R', then the whole household has it.
For this reason, the household level variable is set to 'D' or 'R'
when the person value is 'D' or 'R' respectively, only if the
household level is not already set to 1. */

%MACRO drcheck(hdiff=,pdiff=);
        if &pdiff in ('D','R',' ')
           and &hdiff ne '1'
        then
           do;
              &hdiff = &pdiff;
              dkrfct+1;
        end;
%MEND drcheck;

proc sort data=inpdr.houshld out=hh;
by &msrtkey;
run;

proc sort data=inpdr2.person out=per;
by &msrtkey person;
run;

data hmddata;
merge per (in = pers)  hh (in = house);
by &msrtkey;
if house then output;
run;

data rcd.hmdrcd (keep= &msrtkey HKCAB HCAB HSTOVE HCTRUSE HGETBR
          HSINK HFAUCET HBTUB HWSHWR HSTOOP HREACH HGRASP
          MOBUSE HMWHEELN withdiff dkrfct);
        retain HMWHEELN 0;
        set hmddata;
        by &msrtkey;

        retain HKCAB ' ';
        retain HCAB ' ';
        retain HSTOVE ' ';
        retain HCTRUSE ' ';
        retain HGETBR ' ';
        retain HSINK ' ';
        retain HFAUCET ' ';
        retain HBTUB ' ';
        retain HWSHWR ' ';
        retain HSTOOP ' ';
        retain HREACH ' ';
        retain HGRASP ' ';
        retain withdiff 0;
        retain dkrfct 0;

        if STATUS eq 1 then
        do;
              /* The counters and household level variables are reset for
                each first unique control number. */

                if first.&msrtkey then
                do;
                    HKCAB = '2';
                    HCAB = '2';
                    HSTOVE = '2';
                    HCTRUSE = '2';
                    HGETBR = '2';
                    HSINK = '2';
                    HFAUCET = '2';
                    HBTUB = '2';
                    HWSHWR = '2';
                    HSTOOP = '2';
                    HREACH = '2';
                    HGRASP = '2';
                    withdiff = 0;
                    dkrfct = 0;
                end;

             /* Passing HMD data into the macro checking for people
                   with difficulties. */

                %difcheck(hdiff=HKCAB,pdiff=PKCAB)
                %difcheck(hdiff=HCAB,pdiff=POCAB)
                %difcheck(hdiff=HSTOVE,pdiff=PSTOVE)
                %difcheck(hdiff=HCTRUSE,pdiff=PCTRUSE)
                %difcheck(hdiff=HGETBR,pdiff=PGETBR)
                %difcheck(hdiff=HSINK,pdiff=PSINK)
                %difcheck(hdiff=HFAUCET,pdiff=PFAUCET)
                %difcheck(hdiff=HBTUB,pdiff=PBTUB)
                %difcheck(hdiff=HWSHWR,pdiff=PWSHWR)
                %difcheck(hdiff=HSTOOP,pdiff=PSTOOP)
                %difcheck(hdiff=HREACH,pdiff=PREACH)
                %difcheck(hdiff=HGRASP,pdiff=PGRASP)

                %drcheck(hdiff=HKCAB,pdiff=PKCAB)
                %drcheck(hdiff=HCAB,pdiff=POCAB)
                %drcheck(hdiff=HSTOVE,pdiff=PSTOVE)
                %drcheck(hdiff=HCTRUSE,pdiff=PCTRUSE)
                %drcheck(hdiff=HGETBR,pdiff=PGETBR)
                %drcheck(hdiff=HSINK,pdiff=PSINK)
                %drcheck(hdiff=HFAUCET,pdiff=PFAUCET)
                %drcheck(hdiff=HBTUB,pdiff=PBTUB)
                %drcheck(hdiff=HWSHWR,pdiff=PWSHWR)
                %drcheck(hdiff=HSTOOP,pdiff=PSTOOP)
                %drcheck(hdiff=HREACH,pdiff=PREACH)
                %drcheck(hdiff=HGRASP,pdiff=PGRASP)


   end;    /* end of status check */
   else
   do;   /* status = 2,3,4  */
                if first.&msrtkey then
                do;
                    HKCAB = 'B';
                    HCAB = 'B';
                    HSTOVE = 'B';
                    HCTRUSE = 'B';
                    HGETBR = 'B';
                    HSINK = 'B';
                    HFAUCET = 'B';
                    HBTUB = 'B';
                    HWSHWR = 'B';
                    HSTOOP = 'B';
                    HREACH = 'B';
                    HGRASP = 'B';
                end;
   end;

        if last.&msrtkey then
        do;
         if status in ('2','3','4') then
           MOBUSE = 'B';
         else if status='1' then
         do;
           MOBUSE = '2';
           if WCHAIR in ('D','R',' ') then
              MOBUSE = WCHAIR;
           else
           do;
             if WCHAIR = '1' or
                ECHAIR = '1' or
                CLIFT = '1' or
                CRUTCH = '1' or
                CANE = '1' or
                SPOTHR = '1' then
             MOBUSE = '1';
           end;
         end;

         if status eq '4' then
            HMWHEELN = .B;
         else
         do;
            HMWHEELN = 0;
            if HMOUTLET in ('1','2') then HMWHEELN+1;
            if HMSWITCH in ('1','2') then HMWHEELN+1;
            if HMCLCTRL in ('1','2') then HMWHEELN+1;
            if HMACAB in ('1','2') then HMWHEELN+1;
            if HMCOUNT in ('1','2') then HMWHEELN+1;
            if HMKIT in ('1','2') then HMWHEELN+1;
            if HMBROOM in ('1','2') then HMWHEELN+1;
         end;

         output rcd.hmdrcd;
       end;
run;

